



THE  
POWER  
TO KNOW.

# **SAS/STAT<sup>®</sup> 13.2 User's Guide**

## **The IRT Procedure**

This document is an individual chapter from *SAS/STAT® 13.2 User's Guide*.

The correct bibliographic citation for the complete manual is as follows: SAS Institute Inc. 2014. *SAS/STAT® 13.2 User's Guide*. Cary, NC: SAS Institute Inc.

Copyright © 2014, SAS Institute Inc., Cary, NC, USA

All rights reserved. Produced in the United States of America.

**For a hard-copy book:** No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

**For a Web download or e-book:** Your use of this publication shall be governed by the terms established by the vendor at the time you acquire this publication.

The scanning, uploading, and distribution of this book via the Internet or any other means without the permission of the publisher is illegal and punishable by law. Please purchase only authorized electronic editions and do not participate in or encourage electronic piracy of copyrighted materials. Your support of others' rights is appreciated.

**U.S. Government License Rights; Restricted Rights:** The Software and its documentation is commercial computer software developed at private expense and is provided with RESTRICTED RIGHTS to the United States Government. Use, duplication or disclosure of the Software by the United States Government is subject to the license terms of this Agreement pursuant to, as applicable, FAR 12.212, DFAR 227.7202-1(a), DFAR 227.7202-3(a) and DFAR 227.7202-4 and, to the extent required under U.S. federal law, the minimum restricted rights as set out in FAR 52.227-19 (DEC 2007). If FAR 52.227-19 is applicable, this provision serves as notice under clause (c) thereof and no other notice is required to be affixed to the Software or documentation. The Government's rights in Software and documentation shall be only those set forth in this Agreement.

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

August 2014

SAS provides a complete selection of books and electronic products to help customers use SAS® software to its fullest potential. For more information about our offerings, visit [support.sas.com/bookstore](http://support.sas.com/bookstore) or call 1-800-727-3228.

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.



# Gain Greater Insight into Your SAS<sup>®</sup> Software with SAS Books.

Discover all that you need on your journey to knowledge and empowerment.





# Chapter 53

## The IRT Procedure

### Contents

---

Overview: IRT Procedure . . . . .	<b>4008</b>
Basic Features . . . . .	4008
Getting Started: IRT Procedure . . . . .	<b>4009</b>
Syntax: IRT Procedure . . . . .	<b>4013</b>
PROC IRT Statement . . . . .	4014
BY Statement . . . . .	4022
COV Statement . . . . .	4022
EQUALITY Statement . . . . .	4026
FACTOR Statement . . . . .	4030
FREQ Statement . . . . .	4034
GROUP Statement . . . . .	4034
MODEL Statement . . . . .	4034
VAR Statement . . . . .	4035
VARIANCE Statement . . . . .	4035
WEIGHT Statement . . . . .	4037
Details: IRT Procedure . . . . .	<b>4038</b>
Notation for the Item Response Theory Model . . . . .	4038
Assumptions . . . . .	4039
PROC IRT Contrasted with Other SAS Procedures . . . . .	4039
Response Models . . . . .	4040
Marginal Likelihood . . . . .	4041
Approximating the Marginal Likelihood . . . . .	4041
Maximizing the Marginal Likelihood . . . . .	4042
Factor Score Estimation . . . . .	4044
Model and Item Fit . . . . .	4045
Item and Test information . . . . .	4046
ODS Table Names . . . . .	4046
ODS Graphics . . . . .	4048
Examples: IRT Procedure . . . . .	<b>4048</b>
Example 53.1: Unidimensional IRT Models . . . . .	4048
Example 53.2: Multidimensional Exploratory and Confirmatory IRT Models . . . . .	4063
Example 53.3: Multiple-Group Analysis . . . . .	4067
Example 53.4: Quality of Life Survey . . . . .	4071
References . . . . .	<b>4078</b>

---

## Overview: IRT Procedure

The item response theory (IRT) model was first proposed in the field of psychometrics for the purpose of ability assessment. It is most widely used in education to calibrate and evaluate items in tests, questionnaires, and other instruments and to score subjects on their abilities, attitudes, or other latent traits. Today, all major psychological and educational tests are built using IRT, because the methodology can significantly improve measurement accuracy and reliability while providing potential significant reductions in assessment time and effort, especially via computerized adaptive testing. In a computerized adaptive test, items are optimally selected for each subject. Different subjects might receive entirely different items during the test. IRT plays an essential role in selecting the most appropriate items for each subject and equating scores for subjects who receive different subsets of items. Notable examples of these tests include the Scholastic Aptitude Test (SAT), Graduate Record Examination (GRE), and Graduate Management Admission Test (GMAT). In recent years, IRT models have also become increasingly popular in health behavior, quality of life, and clinical research. The Patient Reported Outcomes Measurement Information System (PROMIS) project, funded by the US National Institutes of Health, is an excellent example. By using IRT, it aims to develop item banks that clinicians and researchers can use to collect important information about therapeutic effects that is not available from traditional clinical measures.

Early IRT models (such as the Rasch model and two-parameter model) concentrate mainly on dichotomous responses. These models were later extended to incorporate other formats, such as ordinal responses, rating scales, partial credit scoring, and multiple category scoring. Early applications of IRT focused primarily on the unidimensional model, which assumes that subject responses are affected only by a single latent trait. Multidimensional IRT models have been developed, but because of their greater complexity, the majority of IRT applications still rely on unidimensional models.

For an introduction to IRT models, see De Ayala (2009) and Embretson and Reise (2000).

---

## Basic Features

The IRT procedure enables you to estimate various item response theory models. The following list summarizes some of the basic features of the IRT procedure:

- uses the Rasch model; one-, two-, three-, and four-parameter models; and graded response model with logistic or probit link
- enables different items to have different response models
- performs multidimensional exploratory and confirmatory analysis
- performs multiple-group analysis, with fixed values and equality constraints within and between groups
- estimates factor scores by using maximum likelihood (ML), maximum a posteriori (MAP), and expected a posteriori (EAP) methods

## Getting Started: IRT Procedure

This example shows how you can use all default settings in PROC IRT to fit an item response model. In this example, there are 50 subjects and each subject responds to 10 items. These 10 items have binary responses: 1 indicates correct and 0 indicates incorrect.

The following DATA step creates the SAS data set IrtBinary:

```
data IrtBinary;
  input item1-item10 @@;
  datalines;
1 0 1 1 1 1 1 1 1 0 1 1 1 1 1 1 1 1 1 0 0 0 0 1 0 1 0 0 1 1 1 1 1 1
0 0 0 0 0 1 1 1 1 1 0 0 0 1 1 1 1 1 1 1 0 0 1 0 1 1 1 1 1 1 1 1 1 1
... more lines ...

1 1 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1 1 0 1 1
;
```

The following statements fit an IRT model:

```
proc irt data=IrtBinary;
  var item1-item10;
run;
```

The PROC IRT statement invokes the procedure, and the **DATA=** option specifies the input data set IrtBinary. The **VAR** statement names the variables to be used in the model. As you can see from the syntax in this example, fitting a IRT model can be very simple when you use the default settings. These default settings are chosen to reflect setups that are common in practice. Some of the important default settings follow:

- The number of factors is 1.
- The two-parameter logistic model is assumed for binary variables, and the graded response model is assumed for ordinal variables.
- The link function is logistic link.
- The estimation method is based on marginal likelihood.
- The optimization method is the quasi-Newton algorithm.
- The quadrature method is adaptive Gauss-Hermite quadrature, in which the number of quadrature points per dimension is determined adaptively.

As a result, the preceding statements fit two-parameter logistic (2PL) models for all the variables that are listed in the **VAR** statement.

The first table that PROC IRT produces is the “Modeling Information” table, as shown in [Figure 53.1](#). This table displays basic information about the analysis, such as the name of the input data set, the link function, the number of items and factors, the number of observations, and the estimation method. You can change the link function by using the **LINK=** option in the **PROC IRT** statement. You can change the response model for all the items by using the **RESFUNC=** option in the **PROC IRT** statement. You can specify different response functions or models for different set of variables by including a **MODEL** statement. If you want

to do multidimensional exploratory analysis, you can simply change the number of factors by using the **NFACTOR=** option in the **PROC IRT** statement. For confirmatory analysis, you can use the **FACTOR** statement to specify the confirmatory factor pattern; the number of factors is implicitly defined by the number of distinctive factor names that you specify in the **FACTOR** statement.

**Figure 53.1** Model Information  
**The IRT Procedure**

Modeling Information	
Data Set	WORK.IRTBINARY
Link Function	Logit
Response Model	Two Parameter Model
Number of Items	10
Number of Factors	1
Number of Observations Read	100
Number of Observations Used	100
Estimation Method	Marginal Maximum Likelihood

The “Item Information” table, shown in [Figure 53.2](#), is displayed by default and can be used to check the item-level information. In this case, each of the 10 variables has two levels, and the raw values for these two levels are 0 and 1, respectively.

**Figure 53.2** Item Information

Item Information		
Item	Levels	Values
item1	2	0 1
item2	2	0 1
item3	2	0 1
item4	2	0 1
item5	2	0 1
item6	2	0 1
item7	2	0 1
item8	2	0 1
item9	2	0 1
item10	2	0 1

The eigenvalues of polychoric correlations are also computed by default and are shown in [Figure 53.3](#). You can use the information from these eigenvalues to assess a reasonable range for the number of factors. For this example, you can observe that the first eigenvalue accounts for almost 50% of the variance, which suggests that there is only one dominant eigenvalue and that a unidimensional model is reasonable for this example. To produce the polychoric correlation table, you specify the **POLYCHORIC** option in the **PROC IRT** statement.



**Figure 53.3** Eigenvalues of Polychoric Correlation

Eigenvalues of the Polychoric Correlation Matrix				
	Eigenvalue	Difference	Proportion	Cumulative
1	4.71105177	3.51149453	0.4711	0.4711
2	1.19955723	0.14183502	0.1200	0.5911
3	1.05772221	0.26577735	0.1058	0.6968
4	0.79194486	0.07204549	0.0792	0.7760
5	0.71989938	0.17782491	0.0720	0.8480
6	0.54207446	0.12713664	0.0542	0.9022
7	0.41493782	0.10631770	0.0415	0.9437
8	0.30862012	0.12256183	0.0309	0.9746
9	0.18605829	0.11792444	0.0186	0.9932
10	0.06813385		0.0068	1.0000

Next, the “Optimization Information” table, shown in [Figure 53.4](#), lists the optimization technique, the numeric quadrature method, and the number of quadrature points per dimension. If you want to use the expectation-maximization (EM) technique, specify **TECHNIQUE=EM** in the **PROC IRT** statement. If you specify the **NOAD** option in the **PROC IRT** statement, PROC IRT uses the nonadaptive Gauss-Hermite quadrature to approximate the likelihood. You can change the number of quadrature points by specifying the **QPOINTS=** option in the **PROC IRT** statement.

**Figure 53.4** Optimization Information

Optimization Information	
Optimization Technique	Quasi-Newton
Likelihood Approximation	Adaptive Gauss-Hermite Quadrature
Number of Quadrature Points	11
Number of Free Parameters	20

[Figure 53.5](#) shows the “Iteration History” table. For each iteration, the table displays the current iteration number, number of function evaluations, objective function value, change of object function value, and maximum value of gradients. You can use this information to monitor the estimation status of the model. You can turn off the display of the “Iteration History” table by specifying the **NOITPRINT** option in the **PROC IRT** statement.

Following the “Iteration History” table is the convergence status table, shown in [Figure 53.6](#). It shows whether the optimization algorithm converges successfully or not. You should make sure that the optimization converges successfully before you try to interpret the estimation results.

**Figure 53.5** Iteration History

Iteration History				
Iteration	Evaluations	Objective Function	Change	Max Gradient
0	2	5.53319820	5.53319820	0.055036
1	4	5.43364722	-0.09955098	0.017483
2	6	5.41529856	-0.01834866	0.017061
3	8	5.40395721	-0.01134135	0.007767
4	10	5.40207771	-0.00187951	0.007106
5	12	5.40147004	-0.00060767	0.003499
6	15	5.40129723	-0.00017281	0.001441
7	18	5.40123492	-0.00006231	0.000867
8	21	5.40120879	-0.00002613	0.000798
9	24	5.40119848	-0.00001031	0.000448
10	27	5.40119212	-0.00000636	0.000145
11	30	5.40119134	-0.00000077	0.000159
12	33	5.40119081	-0.00000054	0.000081
13	36	5.40119060	-0.00000021	0.000045
14	39	5.40119057	-0.00000003	0.00002
15	42	5.40119057	-0.00000001	8.572E-6

**Figure 53.6** Convergence Status

Convergence criterion (GCONV=.000000010) satisfied.
---

Next is the “Model Fit Statistics” table, shown in [Figure 53.7](#), which includes the log likelihood, Akaike’s information criterion (AIC), and the Bayesian information criterion (BIC). If all the response patterns are observed, Pearson’s chi-square and likelihood ratio chi-square statistics are also included in this table. Because some of the response patterns in this example are not observed, the Pearson’s chi-square and likelihood ratio chi-square statistics are not included in the table.

**Figure 53.7** Fit Statistics

Model Fit Statistics	
Log Likelihood	-540.1190565
AIC (Smaller is Better)	1120.238113
BIC (Smaller is Better)	1172.3415168

Finally, the “Item Parameter Estimates” table, shown in [Figure 53.8](#), includes parameter estimates, standard errors, and  $p$ -values. Parameters are organized and displayed within each item. The items are listed in the order of their appearance in the modeling statements. For each item, there are two parameters: difficulty and slope. Difficulty parameters measure the difficulties of the items. As the value of the difficulty parameter increases, the item becomes more difficult. In [Figure 53.8](#), you can observe that all the difficulty parameters are less than 0, which suggests that all the items in this example are relatively easy. The slope parameter values for this example range from 0.94 to 2.33, suggesting that all the items are adequate measures of the latent trait.

**Figure 53.8** Parameter Estimates

Item Parameter Estimates				
Item	Parameter	Estimate	Standard Error	Pr >  t
item1	Difficulty	-0.87121	0.20083	<.0001
	Slope	2.20624	0.67941	0.0006
item2	Difficulty	-1.02199	0.21318	<.0001
	Slope	2.32649	0.76037	0.0011
item3	Difficulty	-0.91668	0.20857	<.0001
	Slope	2.17452	0.68330	0.0007
item4	Difficulty	-0.92919	0.22707	<.0001
	Slope	1.86354	0.57110	0.0006
item5	Difficulty	-1.09791	0.30594	0.0002
	Slope	1.33344	0.42511	0.0009
item6	Difficulty	-0.49151	0.24385	0.0219
	Slope	1.17140	0.36940	0.0008
item7	Difficulty	-0.62129	0.30189	0.0198
	Slope	0.94209	0.32563	0.0019
item8	Difficulty	-0.51111	0.28477	0.0363
	Slope	0.95367	0.32914	0.0019
item9	Difficulty	-0.41404	0.24477	0.0454
	Slope	1.11314	0.35530	0.0009
item10	Difficulty	-0.62670	0.27982	0.0126
	Slope	1.04867	0.34845	0.0013

## Syntax: IRT Procedure

The following statements are available in the IRT procedure:

```

PROC IRT < options > ;
  BY variables ;
  COV covariance parameters ;
  EQUALITY equality-constraints ;
  FACTOR factor-variables-relations ;
  FREQ variable ;
  GROUP variable ;
  MODEL model-specification ;
  VAR variables ;
  VARIANCE variance parameters ;
  WEIGHT variable ;

```

## PROC IRT Statement

**PROC IRT** <options> ;

The PROC IRT statement invokes the IRT procedure. Table 53.1 summarizes the *options* available in the PROC IRT statement. The sections that follow the table describe the PROC IRT statement *options* and then describe the other statements in alphabetical order.

**Table 53.1** PROC IRT Statement Options

Option	Description
<b>Basic Options</b>	
DATA=	Specifies the input data set
DESCENDING	Reverses the sort order of the levels of the response variable
ITEMFIT	Computes the item fit statistics and displays them in a table
LINK=	Specifies the link function
NFACTOR=	Specifies the number of factors
OUT=	Specifies the output data set for factor scores
RESFUNC=	Specifies the response function
RORDER=	Specifies the sort order of the response variables
SCOREMETHOD=	Specifies the factor score estimation method
<b>Computational Options</b>	
ABSFCONV=	Specifies an absolute function difference convergence criterion
ABSGCONV=	Specifies an absolute gradient convergence criterion
ABSPCONV=	Specifies a maximum absolute parameter difference convergence criterion
FCONV=	Specifies a relative function convergence criterion
GCONV=	Specifies a relative gradient convergence criterion
MAXFUNC=	Specifies the maximum number of function calls in the optimization process
MAXITER=	Specifies the maximum number of iterations in the optimization process
MAXMITER=	Specifies the maximum number of iterations in the maximization step of the EM algorithm
NOAD	Specifies nonadaptive quadrature
QPOINTS=	Specifies the number of quadrature points per dimension
TECHNIQUE=	Specifies the optimization technique to obtain maximum likelihood estimates.
<b>Display Options</b>	
NOITPRINT	Suppresses the display of the “Iteration History” table
NOPRINT	Suppresses all ODS output
PINITIAL	Displays initial parameter estimates
POLYCHORIC	Displays the polychoric correlation matrix
PLOTS=	Controls plots that are produced through ODS Graphics
<b>Rotation Method and Properties</b>	
RCONVERGE=	Specifies the convergence criterion for rotation cycles

Table 53.1 *continued*

Option	Description
RITER=	Specifies the maximum number of rotation cycles
ROTATE=	Specifies the rotation method

## PROC IRT Statement Options

**ABSFCONV=***r*

**ABSFTOL=***r*

specifies an absolute function difference convergence criterion. Termination requires a small change of the function value in successive iterations,

$$|f(\boldsymbol{\psi}^{(k-1)}) - f(\boldsymbol{\psi}^{(k)})| \leq r$$

where  $\boldsymbol{\psi}$  denotes the vector of parameters that participate in the optimization and  $f(\cdot)$  is the objective function. This criterion is not used by the expectation-maximization (EM) algorithm. By default,  $r = 0$ .

**ABSGCONV=***r*

**ABSGTOL=***r*

specifies an absolute gradient convergence criterion. Termination requires the maximum absolute gradient element to be small,

$$\max_j |g_j(\boldsymbol{\psi}^{(k)})| \leq r$$

where  $\boldsymbol{\psi}$  denotes the vector of parameters that participate in the optimization and  $g_j(\cdot)$  is the gradient of the objective function with respect to the  $j$ th parameter. This criterion is not used by the EM algorithm. By default,  $r = 1\text{E-}5$ .

**ABSPCONV=***r*

**ABSPTOL=***r*

specifies a maximum absolute parameter difference convergence criterion. This criterion is used only by the EM algorithm. Termination requires the maximum absolute parameter change in successive iterations to be small,

$$\max_j |\boldsymbol{\psi}_j^{(k-1)} - \boldsymbol{\psi}_j^{(k)}| \leq r$$

where  $\boldsymbol{\psi}_j$  denotes the  $j$ th parameter that participates in the optimization. By default,  $r = 1\text{E-}4$ .

**DATA=***SAS-data-set*

specifies the *SAS-data-set* to be read by PROC IRT. The default value is the most recently created data set.

**DESCENDING**

**DESC**

reverses the sorting order for the levels of the response variables. If both the DESCENDING and RORDER= options are specified, PROC IRT orders the levels according to the RORDER= option and then reverses that order.

**FCONV=r****FTOL=r**

specifies a relative function convergence criterion. Termination requires a small relative change of the function value in successive iterations,

$$\frac{|f(\boldsymbol{\psi}^{(k)}) - f(\boldsymbol{\psi}^{(k-1)})|}{|f(\boldsymbol{\psi}^{(k-1)})|} \leq r$$

where  $\boldsymbol{\psi}$  denotes the vector of parameters that participate in the optimization and  $f(\cdot)$  is the objective function. This criterion is not used by the EM algorithm. By default,  $r = 10^{-\text{FDIGITS}}$ , where FDIGITS is, by default,  $-\log_{10}\{\epsilon\}$  and  $\epsilon$  is the machine precision.

**GCONV=r****GTOL=r**

specifies a relative gradient convergence criterion. For all techniques except CONGRA, termination requires the normalized predicted function reduction to be small,

$$\frac{\mathbf{g}(\boldsymbol{\psi}^{(k)})' [\mathbf{H}^{(k)}]^{-1} \mathbf{g}(\boldsymbol{\psi}^{(k)})}{|f(\boldsymbol{\psi}^{(k)})|} \leq r$$

where  $\boldsymbol{\psi}$  denotes the vector of parameters that participate in the optimization,  $f(\cdot)$  is the objective function, and  $\mathbf{g}(\cdot)$  is the gradient. For the CONGRA technique (for which a reliable Hessian estimate  $\mathbf{H}$  is not available), the following criterion is used:

$$\frac{\|\mathbf{g}(\boldsymbol{\psi}^{(k)})\|_2^2 \|\mathbf{s}(\boldsymbol{\psi}^{(k)})\|_2}{\|\mathbf{g}(\boldsymbol{\psi}^{(k)}) - \mathbf{g}(\boldsymbol{\psi}^{(k-1)})\|_2 |f(\boldsymbol{\psi}^{(k)})|} \leq r$$

This criterion is not used by the EM algorithm. By default,  $r = 1\text{E-}8$ .

**ITEMFIT**

displays the item fit statistics. These item fit statistics apply only to binary items that have one latent factor.

**LINK=name**

specifies the link function. You can specify the following values:

**LOGIT** requests the logistic link function.

**PROBIT** requests the probit link function.

By default, LINK=LOGIT.

**MAXFUNC=n****MAXFU=n**

specifies the maximum number of function calls in the optimization process. This option is not used by the EM algorithm. The default values are as follows, depending on which optimization technique is specified in the **TECHNIQUE=** option:

- NRRIDG: 125

- QUANEW: 500
- CONGRA: 1000

The optimization can terminate only after completing a full iteration. Therefore, the number of function calls that are actually performed can exceed the number that is specified by this option.

**MAXITER=*n***

**MAXIT=*n***

specifies the maximum number of iterations in the optimization process. The default values are as follows, depending on which optimization technique is specified in the **TECHNIQUE=** option:

- NRRIDG: 50
- QUANEW: 200
- CONGRA: 400
- EM: 500

**MAXMITER=*n***

**MAXMIT=*n***

specifies the maximum number of iterations in the maximization step of the EM algorithm. By default,  $n = 1$ .

**NFACTOR=*i***

**NFACT=*i***

specifies the number of factors,  $i$ , in the model. You must specify the number of factors only for exploratory analysis, in which all the slope parameters of the items are freely estimated without being explicitly constrained by using the **FACTOR** statement. By default, NFACTOR=1. When you use the **FACTOR** statement to specify the confirmatory factor pattern, the number of factors is implicitly defined by the number of distinctive factor names that you specify in the statement.

**NOAD**

requests that the Gaussian quadrature be nonadaptive.

**NOITPRINT**

suppresses the display of the “Iteration History” table.

**NOPRINT**

suppresses all output displays.

**TECHNIQUE=CONGRA | EM | NONE | NRRIDG | QUANEW**

**TECH=CONGRA | EM | NONE | NRRIDG | QUANEW**

**OMETHOD=CONGRA | EM | NONE | NRRIDG | QUANEW**

specifies the optimization technique to obtain maximum likelihood estimates. You can specify the following techniques:

- |               |   |
|---------------|---|
| <b>CONGRA</b> | performs a conjugate-gradient optimization. |
| <b>EM</b>     | performs an EM optimization.                |
| <b>NONE</b>   | performs no optimization.                   |

**NRRIDG** performs a Newton-Raphson optimization with ridging.  
**QUANEW** performs a dual quasi-Newton optimization.

By default, **TECHNIQUE=QUANEW**.

For more information about these optimization methods (except EM), see the section “[Choosing an Optimization Algorithm](#)” on page 501 in Chapter 19, “[Shared Concepts and Topics](#).” For more information about the EM algorithm, see “Expectation-Maximization (EM) Algorithm” in the section “[Details: IRT Procedure](#)” on page 4038.

**OUT=SAS-data-set**

creates an output data set that contains all the data in the **DATA=** data set plus estimated factor scores. For exploratory analysis, the factor scores are named **\_Factor1**, **\_Factor2**, and so on. For confirmatory analysis, user-specified factor names are used.

PROC IRT provides three estimation methods for factor scores. You can specify the method by using the **SCOREMETHOD** option. The default estimation method, maximum a posteriori (MAP), is used if the **SCOREMETHOD** option is not specified.

**POLYCHORIC**

displays the polychoric correlation matrix.

**PINITIAL**

displays the initial parameter estimates.

**PLOTS** <(global-plot-options)> <= plot-request <(options)>>

**PLOTS** <(global-plot-options)> <= (plot-request <(options)> <...plot-request <(options)>>>

controls the plots that are produced through ODS Graphics. When you specify only one *plot-request*, you can omit the parentheses around it. For example:

```
plots=all
plots=ICC(unpack)
plots(unpack)=(scree ICC)
```

ODS Graphics must be enabled before plots can be requested. For example:

```
ods graphics on;
proc irt plots=all;
run;
ods graphics off;
```

For more information about enabling and disabling ODS Graphics, see the section “[Enabling and Disabling ODS Graphics](#)” on page 606 in Chapter 21, “[Statistical Graphics Using ODS](#).”

You can specify the following *global-plot-option*, which applies to all plots that the IRT procedure generates:



**UNPACK | UNPACKPANEL**

suppresses paneling. By default, multiple plots can appear in some output panels. Specify UNPACK to display each plot individually. You can also specify UNPACK as a suboption in the ICC, IIC, and SCREE options.

You can specify the following *plot-requests*:

**ALL**

displays all default plots.

**ICC <(UNPACK | UNPACKPANEL)>**

displays item characteristic curves (ICCs). By default, multiple ICC plots appear in some output panels. You can request an individual ICC plot for each item by specifying the UNPACK suboption.

**IIC <(UNPACK | UNPACKPANEL)>**

displays item information curves (IICs). By default, multiple IIC plots appear in some output panels. You can request an individual IIC plot for each item by specifying the UNPACK suboption.

**NONE**

suppresses all plots.

**POLYCHORIC <options>****PLCORR <options>**

displays a heat map for the polychoric correlation matrix. You can specify one or both of the following *options*:

**OUTLINE=ON | OFF**

specifies whether to display an outline of the regions in the polychoric correlation heat map. By default, OUTLINE=ON.

**FUZZ= $p$** 

displays polychoric correlations whose absolute values are less than  $p$  as 0 in the heat map. This option is useful when you want to focus on the patterns of sizable correlations that are larger than  $p$  in the heat map. By default, FUZZ = 0.

**SCREE <(UNPACK | UNPACKPANEL)>**

displays the scree and variance explained plots in the same panel. You can display these plots individually by specifying the UNPACK suboption.

**TIC**

displays a test information curve (TIC) plot.

**QPOINTS= $i$** 

specifies the number of quadrature points in each dimension of the integral. If there are  $d$  latent factors and  $n$  quadrature points, the IRT procedure evaluates  $n^d$  conditional log likelihoods for each observation to compute one value of the objective function. Increasing the number of quadrature nodes can substantially increase the computational burden. If you do not specify the number of quadrature points, it is determined adaptively by using the initial parameter estimates.

**RCONVERGE=*p*****RCONV=*p***

specifies the convergence criterion for rotation cycles. Rotation stops when the scaled change of the simplicity function value is less than the RCONVERGE= value. The default convergence criterion is

$$|f_{new} - f_{old}|/K < \epsilon$$

where  $f_{new}$  and  $f_{old}$  are simplicity function values of the current cycle and the previous cycle, respectively;  $K = \max(1, |f_{old}|)$  is a scaling factor; and  $\epsilon$  is 1E-9 by default and is modified by the RCONVERGE= value.

**RESFUNC=ONEP | TWOP | THREEP | FOURP | GRADED | RASCH**

specifies the response functions for the variables that are included in the [VAR](#) statement. The response functions correspond to different response models. You can specify the following values:

<b>ONEP</b>	specifies the one-parameter model.
<b>TWOP</b>	specifies the two-parameter model.
<b>THREEP</b>	specifies the three-parameter model.
<b>FOURP</b>	specifies the four-parameter model.
<b>GRADED</b>	specifies the graded response model.
<b>RASCH</b>	specifies the Rasch model.

By default, RESFUNC=TWOP for binary items and RESFUNC=GRADED for ordinal items. The graded response model assumes that the response variables are ordinal-categorical up to 19 levels. All other models assume binary responses. For more information about these response models, see “Response Models” in the [“Details: IRT Procedure”](#) on page 4038 section.

**RITER=*n***

specifies the maximum number of cycles for factor rotation. The default value is the maximum between 10 times the number of variables and 100.

**ROTATE=*name*****R=*name***

specifies the rotation method.

You can specify the following orthogonal rotation methods:

<b>BIQUARTIMAX   BIQMAX</b>	specifies orthogonal biquartimax rotation.
<b>EQUAMAX   E</b>	specifies orthogonal equamax rotation.
<b>NONE   N</b>	specifies that no rotation be performed, leaving the original orthogonal solution.
<b>PARSIMAX   PA</b>	specifies orthogonal parsimax rotation.
<b>QUARTIMAX   QMAX   Q</b>	specifies orthogonal quartimax rotation.
<b>VARIMAX   V</b>	specifies orthogonal varimax rotation.

You can specify the following oblique rotation methods:

<b>BIQUARTIMIN   BIQMIN</b>	specifies biquartimin rotation.
<b>COVARIMIN   CVMIN</b>	specifies covarimin rotation.
<b>OBBIQUARTIMAX   OBIQMAX</b>	specifies oblique biquartimax rotation.
<b>OBEQUAMAX   OE</b>	specifies oblique equamax rotation.
<b>OBPARSIMAX   OPA</b>	specifies oblique parsimax rotation.
<b>OBQUARTIMAX   OQMAX</b>	specifies oblique quartimax rotation.
<b>OBVARIMAX   OV</b>	specifies oblique varimax rotation.
<b>QUARTIMIN   QMIN</b>	specifies quartimin rotation.

By default, ROTATE=VARIMAX.

#### **RORDER=DATA | FORMATTED | FREQ | INTERNAL**

specifies the sort order for the levels of the response variable. This order determines which threshold parameter in the model corresponds to each level in the data. If RORDER=FORMATTED for numeric variables for which you have supplied no explicit format, the levels are ordered by their internal values. This option applies to all the responses in the model. When the default, RORDER=FORMATTED, is in effect for numeric variables for which you have supplied no explicit format, the levels are ordered by their internal values. You can specify the following sort orders:

<b>Value of RORDER=</b>	<b>Levels Sorted By</b>
DATA	Order of appearance in the input data set
FORMATTED	External formatted value, except for numeric variables that have no explicit format, which are sorted by their unformatted (internal) value
FREQ	Descending frequency count; levels that contain the most observations come first in the order
INTERNAL	Unformatted value

For FORMATTED and INTERNAL, the sort order is machine-dependent. For more information about sort order, see the chapter on the SORT procedure in the *SAS Procedures Guide* and the discussion of BY-group processing in *SAS Language Reference: Concepts*.

#### **SCOREMETHOD=ML | EAP | MAP**

specifies the method of factor score estimation. You can specify the following methods:

<b>ML</b>	requests the maximum likelihood method.
<b>EAP</b>	requests the expected a posteriori method.
<b>MAP</b>	requests the maximum a posteriori method.

---

## BY Statement

**BY** *variables* ;

You can specify a BY statement with PROC IRT to obtain separate analyses of observations in groups that are defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables. If you specify more than one BY statement, only the last one specified is used.

If your input data set is not sorted in ascending order, use one of the following alternatives:

- Sort the data by using the SORT procedure with a similar BY statement.
- Specify the NOTSORTED or DESCENDING option in the BY statement for the IRT procedure. The NOTSORTED option does not mean that the data are unsorted but rather that the data are arranged in groups (according to values of the BY variables) and that these groups are not necessarily in alphabetical or increasing numeric order.
- Create an index on the BY variables by using the DATASETS procedure (in Base SAS software).

Because sorting the data changes the order in which PROC IRT reads observations, the sort order for the levels of the response variables might be affected if you also specify **RORDER=DATA** in the **PROC IRT** statement.

For more information about BY-group processing, see the discussion in *SAS Language Reference: Concepts*. For more information about the DATASETS procedure, see the discussion in the *Base SAS Procedures Guide*.

---

## COV Statement

**COV** *assignment* <, *assignment* ... > ;

where *assignment* represents

*var-list* < \* *var-list2* > < = *parameter-spec* >

The COV statement defines the factor covariances in confirmatory models. In each *assignment* of the COV statement, you specify variables in the *var-list* and *var-list2* lists, followed by the covariance parameter specification in the *parameter-spec* list. The last two specifications are optional.

You can specify the following five types of the parameters for the covariances:

- an unnamed free parameter
- an initial value
- a fixed value
- a free parameter with a name provided
- a free parameter with a name and initial value provided

Consider a multidimensional model that has the latent factors FACTOR1, FACTOR2, FACTOR3, and FACTOR4. The following COV statement shows the five types of specifications in five *assignments*:


```
cov FACTOR2 FACTOR1 ,
  FACTOR3 FACTOR1 = (0.3) ,
  FACTOR3 FACTOR2 = 1.0 ,
  FACTOR4 FACTOR1 = phi1 ,
  FACTOR4 FACTOR2 = phi2(0.2) ;
```

In this statement, `cov(FACTOR2, FACTOR1)` is specified as an unnamed free parameter, `cov(FACTOR3, FACTOR1)` is an unnamed free parameter but with an initial value of 0.3, and `cov(FACTOR3, FACTOR2)` is a fixed value of 1.0. This value stays the same in the estimation. `cov(FACTOR4, FACTOR1)` is a free parameter named phi1, and `cov(FACTOR4, FACTOR2)` is a free parameter named phi2 that has an initial value of 0.2.

Note that the *var-list* and *var-list2* lists to the left of the equal sign in the COV statement should contain only names of latent factors that are specified in the **FACTOR** statement.

If you specify only the *var-list* list, then you are specifying the so-called within-list covariances. If you specify both the *var-list* and *var-list2* lists, then you are specifying the so-called between-list covariances. An asterisk is used to separate the two variable lists. You can use one of these two alternatives to specify the covariance parameters. Figure 53.9 illustrates the within-list and between-list covariance specifications.

**Figure 53.9** Within-List and Between-List Covariances



### Within-List Covariances

The left panel of Figure 53.9 shows that the same set of four factors is used in both the rows and the columns. This yields six nonredundant covariances (variances are not included) to specify. In general, for a *var-list* list that has  $k$  variables in the COV statement, you can specify  $k(k - 1)/2$  distinct covariance parameters. The variable order of the *var-list* list is important. For example, the left panel of Figure 53.9 corresponds to the following COV statement:

```
cov F1-F4 = phi1-phi6;
```

This statement is equivalent to the following statement:

```
cov F2 F1 = phi1,
    F3 F1 = phi2, F3 F2 = phi3,
    F4 F1 = phi4, F4 F2 = phi5, F4 F3 = phi6;
```

Another way to assign distinct parameter names that have the same prefix is to use the so-called prefix name. For example, the following COV statement is exactly the same as the preceding statement:

```
cov F1-F4 = 6*phi__ ; /* phi with two trailing underscores */
```

In the COV statement, `phi__` is a prefix name that has the root `phi`. The notation `6*` means that this prefix name is applied six times, resulting in a generation of the six parameter names `phi1`, `phi2`, ..., `phi6` for the six covariance parameters.

The root of the prefix name should have only a few characters so that the generated parameter name is not longer than 32 characters. To avoid unintentional equality constraints, the prefix names should not conflict with other parameter names.

You can also specify the within-list covariances as unnamed free parameters, as shown in the following statement:

```
cov F1-F4;
```

This statement is equivalent to the following statement:

```
cov F2 F1,
    F3 F1, F3 F2,
    F4 F1, F4 F2, F4 F3;
```

## Between-List Covariances

The right panel of [Figure 53.9](#) illustrates the application of the between-list covariance specification. The set of row variables is different from the set of column variables. You intend to specify the cross covariances of the two sets of variables. There are four of these covariances in the figure. In general, for  $k_1$  and  $k_2$  variable names in the two variable lists (separated by an asterisk) in a COV statement, there are  $k_1 \times k_2$  distinct covariances to specify. Again, variable order is very important. For example, the right panel of [Figure 53.9](#) corresponds to the following between-list covariance specification:

```
cov F1 F2 * F3 F4 = phi1-phi4;
```

This is equivalent to the following statement:

```
cov F1 F3 = phi1, F1 F4 = phi2,
    F2 F3 = phi3, F2 F4 = phi4;
```

You can also use the prefix name specification for the same specification, as shown in the following statement:

```
cov F1 F2 * F3 F4 = 4*phi__ ; /* phi with two trailing underscores */
```

## Mixed Parameter Lists

You can specify different types of parameters for the list of covariances. For example, you use a list of parameters that have mixed types in the following statement:

```
cov F1-F4 = phi1(0.1) 0.2 phi3 phi4(0.4) (0.5) phi6;
```

This statement is equivalent to the following statement:

```
cov F2 F1 = phi1(0.1) ,
    F3 F1 = 0.2      , F3 F2 = phi3,
    F4 F1 = phi4(0.4) , F4 F2 = (0.5) , F4 F3 = phi6;
```

Notice that an initial value that follows a parameter name is associated with the free parameter. Therefore, in the original mixed list specification, 0.1 is interpreted as the initial value of the parameter phi1, but not as the initial estimate of the covariance between F3 and F1. Similarly, 0.4 is the initial value of the parameter phi4, but not the initial estimate of the covariance between F4 and F2.

However, if you indeed want to specify that phi1 is a free parameter *without* an initial value and 0.1 is an initial estimate of the covariance between F3 and F1 (while keeping all other things the same), you can use a null initial value specification for the parameter phi1, as shown in the following statement:

```
cov F1-F4 = phi1() (0.1) phi3 phi4(0.4) (0.5) phi6;
```

This way, 0.1 becomes the initial estimate of the covariance between F3 and F1. Because a parameter list that has mixed types might be confusing, you can break down the specifications into separate *assignments* to remove ambiguities. For example, you can use the following equivalent statement:

```
cov F2 F1 = phi1 ,
    F3 F1 = (0.1)      , F3 F2 = phi3,
    F4 F1 = phi4(0.4) , F4 F2 = (0.5) , F4 F3 = phi6;
```

## Shorter and Longer Parameter Lists

If you provide fewer parameters than the number of covariances in the variable lists, all the remaining parameters are treated as unnamed free parameters. For example, the following statement assigns a fixed value to **cov(F1,F3)** while treating all the other three covariances as unnamed free parameters:

```
cov F1 F2 * F3 F4 = 1.0;
```

This statement is equivalent to the following statement:

```
cov F1 F3 = 1.0, F1 F4, F2 F3, F2 F4;
```

If you intend to fill up all values by the last parameter specification in the list, you can use the continuation syntax [...], [...], or [...], as in the following example:

```
cov F1 F2 * F3 F4 = 1.0 phi [...];
```

This means that **cov(F1,F3)** is a fixed value of 1 and all the remaining three covariances are free parameters named phi. The last three covariances are thus constrained to be equal by having the same parameter name.

However, you must be careful not to provide too many parameters. For example, the following statement results in an error:

```
cov F1 F2 * F3 F4 = 1.0 phi2(2.0) phi3 phi4 phi5 phi6;
```

The parameters after phi4 are excessive.

## Default Covariance Parameters

In exploratory analysis, all factor covariances are fixed at zero for the unrotated or orthogonally rotated solutions. For confirmatory analysis, by default all factor covariances are free parameters. You can also use the COV statement to override these default covariance parameters in situations where you want to set parameter constraints or provide initial or fixed values.

---

## EQUALITY Statement

**EQUALITY | EQCON** *equality-constraints* <, *equality-constraints* ... >;

where *equality-constraints* is defined as

*variable-list* < / *constraint-options* >

The EQUALITY statement provides a versatile way to specify various types of equality constraints on the parameters in the model. You can specify within-group or between-group equality constraints on specific sets of parameters for particular sets of variables or factors. In the *variable-list*, you specify the set of variables that are subject to the equality constraints on their respective parameters. You can either specify the names of the variables or use one of the support keywords (see list later in this section) for *variable-list*. In the *constraint-options*, you specify the types of parameters, the specific groups (in multiple-group analysis), and the specific factors (in multidimensional models) on which the equality constraints are imposed.

For example, the following statements specify that all related parameters of x1 through x5 are constrained to be equal:

```
proc irt;
  model x1-x10/resfunc=graded;
  equality x1-x5;
run;
```

Because all items are fitted by the graded response model, all slopes for variables x1–x5 are constrained to be the same and the intercepts for variables x1–x5 are also constrained to be the same. For example, if each of these variables has five categories, there would be four set of constraints, respectively, for each of the four intercept parameters over the five variables.

You can limit the set of parameters for the equality constraints by specifying the PARM= option (one of the *constraint-options*). For example, the following statements constrain only the slope parameters of x1–x5, instead of all related parameters in the graded response model:

```
proc irt;
  model x1-x10/resfunc=graded;
  equality x1-x5/parm=[slopes];
run;
```

There are various ways to specify the target set of variables that are subject to the equality constraints. You can specify variables directly, or you can specify the following *variable-list*:



**\_ALL\_**

specifies all variables and factors in the analysis. Equality constraints on parameters related to factors apply to multiple-group analysis only.

**\_ALLITEM\_**

specifies all variables in the analysis.

**\_ALLONEP\_****\_ALLONEPITEM\_**

specifies all variables that are fitted by the one-parameter model in the analysis.

**\_ALLTWOP\_****\_ALLTWOPITEM\_**

specifies all variables that are fitted by the two-parameter model in the analysis.

**\_ALLTHREEP\_****\_ALLTHREEPITEM\_**

specifies all variables that are fitted by the three-parameter model in the analysis.

**\_ALLFOURP\_****\_ALLFOURPITEM\_**

specifies all variables that are fitted by the four-parameter model in the analysis.

**\_ALLGR\_****\_ALLGRITEM\_**

specifies all variables that are fitted by the graded response model in the analysis.

**\_ALLRASCH\_****\_ALLRASCHITEM\_**

specifies all variables that are fitted by the Rasch model in the analysis.

You can also specify the following keywords, with a list of *excluded-variables* for *variable-list*:

**\_ALL\_BUT\_ [excluded-variables]**

specifies all variables and factors except the *excluded-variables* in the analysis.

**\_ALLITEM\_BUT\_ [excluded-variables]**

specifies all variables except the *excluded-variables* in the analysis.

**\_ALLONEP\_BUT\_ [ excluded-variables ]****\_ALLONEPITEM\_BUT\_ [ excluded-variables ]**

specifies all variables except the *excluded-variables* that are fitted by the one-parameter model in the analysis.

**\_ALLTWOP\_BUT\_ [ excluded-variables ]****\_ALLTWOPITEM\_BUT\_ [ excluded-variables ]**

specifies all variables except the *excluded-variables* that are fitted by the two-parameter model in the analysis.

**`_ALLTHREEP_BUT_`** [ *excluded-variables* ]

**`_ALLTHREEPITEM_BUT_`** [ *excluded-variables* ]

specifies all variables except the *excluded-variables* that are fitted by the three-parameter model in the analysis.

**`_ALLFOURP_BUT_`** [ *excluded-variables* ]

**`_ALLFOURPITEM_BUT_`** [ *excluded-variables* ]

specifies all variables except the *excluded-variables* that are fitted by the four-parameter model in the analysis.

**`_ALLGR_BUT_`** [ *excluded-variables* ]

**`_ALLGRITEM_BUT_`** [ *excluded-variables* ]

specifies all variables except the *excluded-variables* that are fitted by the graded response model in the analysis.

**`_ALLRASCH_BUT_`** [ *excluded-variables* ]

**`_ALLRASCHITEM_BUT_`** [ *excluded-variables* ]

specifies all variables except the *excluded-variables* that are fitted by the Rasch model in the analysis.

For example, if you have mixed model types for the item responses, the equality constraints might be set on a particular set of response variables. The following example shows that the equality constraints are applied to those variables that are fitted by the three-parameter model (that is, x7–x10):

```
proc irt;
  model x1-x6/resfunc=graded,
        x7-x10/resfunc=threep;
  equality _allthreep_;
run;
```

Suppose that the preceding model does not fit well and you want to consider a less restricted model in which the equality constraints are imposed on all variables except x10 in the three-parameter model. The following statements achieve this purpose:

```
proc irt;
  model x1-x6/resfunc=graded,
        x7-x10/resfunc=threep;
  equality _allthreep_but_(x10);
run;
```

In the *constraint-options*, you can specify options for parameter types (PARM= option), the set of groups (BETWEEN\_GP= and WITHIN\_GP= options), and the set of factors. If you do not use these options, all related parameter types, all groups, and all factors are subject to the constraints for the specified set of variables. You can specify the following *constraint-options*:

**BET < =** [ *group-list* ] >

**BETWEEN < =** [ *group-list* ] >

**BETWEEN\_GP < =** [ *group-list* ] >

specifies that the equality constraints be applied across or between groups in the multiple-group analysis. Setting between-group constraints is the default when you fit multiple-group models. Hence, it is not necessary to use this option when you want to set equality constraints between all groups. When only a subset of groups is subject to the intended constraints, you can specify the groups in the *group-list*. This option has no effect if you have only one group in the analysis.

**PARM** < = [ *parameter-types* ] >

specifies the particular types of parameters that are subject to equality constraints. By default, all related parameters are subject to the constraints. You can specify the following *parameter-types*:

**CEIL**

**CEILING**

indicates that the ceiling parameters are constrained.

**GUESS**

**GUESSING**

indicates that the guessing parameters are constrained.

**INTERCEPT**

indicates that the intercept parameters are constrained.

**SLOPE** < [ *factor-list* ] >

**DISCRIMINATION** < [ *factor-list* ] >

indicates that the slope or discrimination parameters are constrained. The optional *factor-list* indicates the set of factors to which the constrained slope parameters pertain. The use of *factor-list* is relevant only when you conduct a confirmatory analysis by specifying the factor pattern in the **FACTOR** statement.

**WIT** < = [ *group-list* ] >

**WITHIN** < = [ *group-list* ] >

**WITHIN\_GP** < = [ *group-list* ] >

specifies that the equality constraints be applied within groups in multiple-group analyses. Setting within-group constraints is the default when you fit a single-group model. Hence, this option is not necessary when you have only one group in the analysis. In multiple-group analyses, between-group constraints are set by default. When you specify this option, within-group constraints are set instead of between-group constraints. You can also specify the specific groups in the *group-list* that are subject to the within-group constraints. The default is to apply the equality constraints to all groups.

You can combine the *constraint-options* to set various types of constraints for your model. You can also specify more than one constraint in an EQUALITY statement. You can even use multiple EQUALITY statements for better organization of the constraints.

For example, suppose that a single-group analysis is conducted using three different types of models (two-parameter, graded responses, and three-parameter model) for the response variables. Consider the following statements:

```
proc irt;
  model x1-x10/resfunc=twop, x11-x20/resfunc=graded, x21-x30/resfunc=threep;
  equality _alltwop_but_(x9-x10),
          x11-x25 / parm=[slope],
          _allthreep_ / parm=[guess];
run;
```

The first set of equality constraints applies to the intercept and slope parameters of x1–x8, leaving the parameters of x9 and x10 freely estimated. The second set of equality constraints applies to the slope parameters of variables x11–x25, even though x21–x25 have a different model type than x11–x20. The

third set of equality constraints applies to the guessing parameters of all variables that are fitted by the three-parameter model (that is, x21–x30).

In multiple-group analysis, constraints are set across groups by default. But within-group constraints can also be set by using the WITHIN\_GP option. Suppose there are three groups in the analysis and the grouping variable GP has three distinct values, 1, 2, and 3. Consider the following example:

```
proc irt;
  group GP;
  model x1-x10/resfunc=twop,
        x11-x20/resfunc=graded,
        x21-x30/resfunc=threep;
  equality _alltwop_but_(x9-x10),
        x11-x25 / parm=[slope] between_gp=[1 2],
        _allthreep_ / parm=[guess] within_gp=[1 3];
run;
```

This example is quite similar to the preceding example, but with some modifications from using the BETWEEN\_GP and WITHIN\_GP options.

The first set of equality constraints is specified exactly the same way as in the preceding example. However, the effect is much different. In the current multiple-group example, the specification constrains the parameters across groups by default. This means that the intercept and slope parameters of x1–x8 are constrained over the three groups. So there would be 16 sets of equality constraints, respectively, for the 16 parameters in variables over the three groups. However, if you use the WITHIN\_GP option, the parameters for x1–x8 are the same within each group. This results in three separate sets of equality constraints on 16 intercept parameters, respectively, for the three groups. Moreover, if you use the WITHIN\_GP and BETWEEN\_GP options together, all 48 parameters in the groups are constrained to be the same.

The second set of equality constraints applies to the slope or discrimination parameters of variables x11–x25 across groups 1 and 2 only, but not all groups. This means that there are 15 equality constraints, respectively, for the 15 slope or discrimination parameters in variables across groups 1 and 2. The discrimination parameters for these variables are not constrained within groups.

The third set of equality constraints applies to the guessing parameters of variables x21–x30 (that is, all the variables that are fitted by the three-parameter model) within groups 1 and 3, respectively. The guessing parameters for these variables are not constrained across groups.

---

## FACTOR Statement

**FACTOR** *factor-variables-relation* <, *factor-variables-relation* ... > ;

where each *factor-variables-relation* is defined as

*factor right-arrow var-list* < = *parameter-spec* >

where *factor* is a name that represents an intended factor; *right-arrow* is ==>, --->, ==>, -->, =>, ->, or >; *var-list* is a list of variables that have nonzero slopes associated with the factor; and *parameter-spec* represents the specifications of parameter name and values (fixed or initial).

You use the FACTOR statement to specify the pattern of relationships between variables and factors in confirmatory models. You do not need to use the FACTOR statement if you are fitting an exploratory model. To complete the specification of a confirmatory model, you might need to use the [VARIANCE](#) and [COV](#) statements to specify the variance and covariance parameters in the model, as shown in the following syntax:

```
FACTOR factor-variable-relation <, factor-variables-relation ... > ;
VARIANCE variance-parameters ;
COV covariance-parameters ;
```

The specifications in the FACTOR statement concern the pattern in the slope matrix. More details follow after a brief description of the [VARIANCE](#) and [COV](#) statements.

By default, the factor variances are fix parameters with a value of 1 in the confirmatory factor model. However, you can override these default parameters by specifying them explicitly in the [VARIANCE](#) statement. For example, in some confirmatory factor models, you might want to free some of these factor variances, or you might want to set equality constraints by using the same parameter name at different parameter locations in your model. Note that if you free some of the factor variances, you need to fix some slope parameters to identify the model.

By default, factor covariances are free parameters in the confirmatory model. However, you can override these default covariance parameters by specifying them explicitly in the [COV](#) statement.

Because the default parameterization of the confirmatory model already covers most commonly used parameters, the specifications in the [VARIANCE](#) and [COV](#) statements are secondary to the specifications in the FACTOR statement, which specifies the pattern of the slope matrix. The following example statement illustrate the syntax of the confirmatory FACTOR statement. Suppose there are nine variables, V1–V9, in your sample and you want to fit a confirmatory IRT model with four factors, as follows:

```
factor
  g_factor    ===>  V1-V9 ,
  factor_a    ===>  V1-V3 ,
  factor_b    ===>  V4-V6 ,
  factor_c    ===>  V7-V9 ;
```

In this factor model, you assume a general factor, `g_factor`, and three group factors, `factor_a`, `factor_b`, and `factor_c`. The general factor, `g_factor`, is related to all variables in the sample, whereas each group factor is related to only three variables. This example fits the following pattern of the slope matrix:

	<code>g_factor</code>	<code>factor_a</code>	<code>factor_b</code>	<code>factor_c</code>
V1	x	x		
V2	x	x		
V3	x	x		
V4	x		x	
V5	x		x	
V6	x		x	
V7	x			x
V8	x			x
V9	x			x

Here an x represents an unnamed free parameter, and all other cells that are blank are fixed zeros.

You can specify the following five types of parameters (*parameter-spec*) at the end of each *factor-variables-relation*:

- an unnamed free parameter
- an initial value
- a fixed value
- a free parameter with a name provided
- a free parameter with a name and initial value provided

To illustrate these different types of parameter specifications, consider the following pattern of slopes:

	<b>g_factor</b>	<b>factor_a</b>	<b>factor_b</b>	<b>factor_c</b>
<b>V1</b>	<b>g_load1</b>	<b>1.</b>		
<b>V2</b>	<b>g_load2</b>	<b>x</b>		
<b>V3</b>	<b>g_load3</b>	<b>x</b>		
<b>V4</b>	<b>g_load4</b>		<b>1.</b>	
<b>V5</b>	<b>g_load5</b>		<b>load_a</b>	
<b>V6</b>	<b>g_load6</b>		<b>load_b</b>	
<b>V7</b>	<b>g_load7</b>			<b>1.</b>
<b>V8</b>	<b>g_load8</b>			<b>load_c</b>
<b>V9</b>	<b>g_load9</b>			<b>load_c</b>

Here an x represents an unnamed free parameter, a constant 1 represents a fixed value, and each name in a cell represents a name for a free parameter. You can specify this pattern by using the following FACTOR statement:

```
factor
  g_factor  ==>  V1-V9    = g_load1-g_load9 (9*0.6) ,
  factor_a  ==>  V1-V3    = 1. (.7 .8) ,
  factor_b  ==>  V4-V6    = 1. load_a (.9) load_b ,
  factor_c  ==>  V7-V9    = 1. 2*load_c ;
```

In the first entry of the FACTOR statement, you specify that the slopes of V1–V9 on g\_factor are the free parameters g\_load1–g\_load9, all of which are given an initial estimate of 0.6. The syntax 9\*0.6 means that 0.6 is repeated nine times. Because they are enclosed in parentheses, all these values are treated as initial estimates but not as fixed values.

You can split the second entry of the FACTOR statement into the following specification:

```
factor_a  ==>  V1    = 1. ,
factor_a  ==>  V2    = (.7) ,
factor_a  ==>  V3    = (.8) ,
```

This means that the first slope is a fixed value of 1 and that the other slopes are unnamed free parameters that have initial estimates of 0.7 and 0.8, respectively.

You can split the third entry of the FACTOR statement into the following specification:

```
factor_b    ==>    V4      = 1.  ,
factor_b    ==>    V5      = load_a (.9) ,
factor_b    ==>    V6      = load_b,
```

This means that the first slope is a fixed value of 1, the second slope is a free parameter named `load_a` with an initial estimate of 0.9, and the third slope is a free parameter named `load_b` without an initial estimate. PROC IRT generates the initial value of this free parameter.

The fourth entry of the FACTOR statement states that the first slope is a fixed 1 and the remaining two slopes are free parameters named `load_c`. No initial estimate is given. But because the two slopes have the same parameter name, they are constrained to be equal in the estimation.

Notice that an initial value that follows a parameter name is associated with the free parameter. For example, in the third entry of the FACTOR statement, the specification `(.9)` after `load_a` is interpreted as the initial value of the parameter `load_a`, but not as the initial estimate of the next slope of `V6`.

However, if you indeed want to specify that `load_a` is a free parameter *without* an initial value and `(0.9)` is an initial estimate for the slope of `V6`, you can use a null initial value specification for the parameter `load_a`, as shown in the following specification:

```
factor_b    ==>    V4-V6    = 1. load_a() (.9) ,
```

This way, 0.9 becomes the initial estimate of the slope of `V6`. Because a parameter list that contains mixed parameter types might be confusing, you can split the specification into separate entries to remove ambiguities. For example, you can use the following equivalent specification:

```
factor_b    ==>    V4      = 1. ,
factor_b    ==>    V5      = load_a ,
factor_b    ==>    V6      = (.9) ,
```

## Shorter and Longer Parameter Lists

If you provide fewer parameters than the number of slopes that are specified in the corresponding *factor-variable-relation*, all the remaining parameters are treated as unnamed free parameters. For example, the following statement assigns a fixed value of 1.0 to the first slope, while treating the remaining two slopes as unnamed free parameters:

```
factor
  factor_a    ==>    V1-V3    = 1. ;
```

This statement is equivalent to the following statement:

```
factor
  factor_a    ==>    V1      = 1. ,
  factor_a    ==>    V2 V3    ;
```

If you intend to fill up all values with the last parameter specification in the list, you can use the continuation syntax `[...]`, `[. .]`, or `[.]`, as shown in the following example:

```
factor
  g_factor    ==>    V1-V30    = 1. (.5) [...];
```

This means that the slope of `V1` on `g_factor` is a fixed value of 1.0 and that the remaining 29 slopes are unnamed free parameters, all of which are given an initial estimate of 0.5.

However, you must be careful not to provide too many parameters. For example, the following statement results in an error:

```
factor
  g_factor    ==>    V1-V3    = load1-load6;
```

The parameter list has six parameters for three slopes. Parameters after load3 are excessive.

## Default Parameters

It is important to understand the default parameters in the FACTOR model. First, if you know which parameters are default free parameters, you can make your specification more efficient by omitting the specifications of those parameters that can be set by default.

---

## FREQ Statement

**FREQ** *variable* ;

If one variable in your data set represents the frequency of occurrence for the other values in the observation, specify the variable's name in a FREQ statement. PROC IRT then treats the data set as if each observation appears  $n_i$  times, where  $n_i$  is the value of the FREQ variable for observation  $i$ . Only the integer portion of the value is used. If the value of the FREQ variable is less than 1 or is missing, that observation is not included in the analysis. The total number of observations is considered to be the sum of the FREQ values.

---

## GROUP Statement

**GROUP** *variable* ;

The GROUP statement specifies the grouping variable that defines the groups of the observations. This statement is required if you intend to do a multiple-group analysis. The values of the grouping variable can be either integers or character strings. PROC IRT analyzes the input data set and determines the number of distinct groups in the data set by counting the number of distinct values in the grouping variable. Because there is no other explicit way to specify the number of groups or the grouping values, you must make sure that all (and only) the intended groups have been indexed properly by the grouping variable in the data set for a multiple-group analysis.

---

## MODEL Statement

**MODEL** *model-specification* <, *model-specification* ... > ;

where *model-specification* is defined as

*variable-list* < / *model-option* >

The MODEL statement specifies the items and their response functions or models. You can specify different response models for different items. In the *variable-list*, you specify the set of variables that use the same model.



You can specify the following *model-option*:

**RESFUNC** < = [ *response-model-types* ] >

specifies the response function or model. For available keywords, see the **RESFUNC=** option in the **PROC IRT** statement. For technical details about these response models, see “Response Models” in the section “[Details: IRT Procedure](#)” on page 4038.

You can specify mixed response models for different items as follows:

```
proc irt;
  model x1-x10/resfunc=twop, x11-x20/resfunc=graded, x21-x30/resfunc=threep;
run;
```

For variables that are also listed in the VAR statement, the model that is specified here overwrites the default model or the model that is specified by using the **RESFUNC=** option in the **PROC IRT** statement.

You can use the **EQUALITY** statement to set equality constraints on these parameters.

---

## VAR Statement

**VAR** | **VARIABLE** *variables* ;

The VAR statement lists the analysis variables or items in the model. If you do a multiple-group analysis, the same set of analysis variables is assumed for all groups. By default, all variables that you specify in the VAR statement are fitted by the graded response model, which assumes that the analysis variables are ordinal and have 2 to 19 levels.

You can overwrite the default response model for the analysis variables in the VAR statement by using the **RESFUNC=** option in the **PROC IRT** statement. If you want to analyze the data by using a mixed type of response model, you can use the **MODEL** statement.

---

## VARIANCE Statement

**VARIANCE** *assignment* < , *assignment* ... > ;

where *assignment* represents

*var-list* < =*parameter-spec*>

The VARIANCE statement specifies the factor variance parameters in connection with the **FACTOR** statement. Notice that the VARIANCE statement is different from the **VAR** statement, which specifies variables for analysis. You can list factors only in the *var-list* of the VARIANCE statement.

In each *assignment* of the VARIANCE statement, you include the *var-list* whose variances you want to specify. Optionally, you can provide a list of parameter specifications (*parameter-spec*) after an equal sign for each *var-list*.

You can specify the following five types of the parameters for the variances of the latent factor in the VARIANCE statement:

- an unnamed free parameter

- an initial value
- a fixed value
- a free parameter with a name provided
- a free parameter with a name and initial value provided

Consider a confirmatory model that has the latent factors F1, F2, F3, F4, and F5.

The following VARIANCE statement illustrates the five types of parameter specifications in five *assignments*:

```
variance
  F1 ,
  F2 = (.5) ,
  F3 = 1.0 ,
  F4 = fvar ,
  F5 = fvar(0.7) ;
```

In this statement, the variance of F1 is specified as an unnamed free parameter. The variance of F2 is an unnamed free parameter that has an initial value of 0.5. The variance of F3 is a fixed value of 1.0. This value stays the same during the estimation. The variance of F4 is a free parameter named fvar1. The variance of F5 is a free parameter named fvar2 that has an initial value of 0.7.

## Mixed Parameter Lists

You can specify different types of parameters for the list of variances. For example, the following statement uses a list of parameters that have mixed types:

```
variance
  F1-F6 = vp1 vp2(2.0) vp3 4. (.3) vp6(.4) ;
```

This is equivalent to the following statement:

```
variance
  F1 = vp1
  F2 = vp2(2.0) ,
  F3 = vp3 ,
  F4 = 4. ,
  F5 = (.3) ,
  F6 = vp6(.4) ;
```

As you can see, an initial value that follows a parameter name is associated with the free parameter. For example, in the original mixed list specification, the specification **(2.0)** after **vp2** is interpreted as the initial value of the parameter **vp2**, but not as the initial estimate of the variance of **F3**.

However, if you indeed want to specify that **vp2** is a free parameter *without* an initial value and 2.0 is an initial estimate of the variance of **F3** (while keeping all other things the same), you can use a null initial value specification for the parameter **vp2**, as shown in the following statement:

```
variance
  F1-F6 = vp1 vp2() (2.0) 4. (.3) vp6(.4) ;
```

This way, 2.0 becomes the initial estimate of the variance of F3. Because a parameter list that contains mixed parameter types might be confusing, you can break down the specifications into separate *assignments* to remove ambiguities. For example, you can use the following equivalent statement:

```
variance
  F1 = vp1
  F2 = vp2,
  F3 = (2.),
  F4 = 4. ,
  F5 = (.3),
  F6 = vp6(.4);
```

### Shorter and Longer Parameter Lists

If you provide fewer parameters than the number of variances in the *var-list*, all the remaining parameters are treated as unnamed free parameters. For example, the following statement assigns a fixed value of 1.0 to the variance of F1 while treating the other three variances as unnamed free parameters:

```
variance
  F1-F4 = 1.0;
```

This statement is equivalent to the following statement:

```
variance
  F1 = 1.0, F2-F4;
```

If you intend to fill up all values with the last parameter specification in the list, you can use the continuation syntax [...], [...], or [...], as shown in the following example:

```
variance
  F1-F100 = 1.0 psi [...];
```

This means that the variance of F1 is fixed at 1.0 and that the variances of F1–F100 are all free parameters named psi. All variances except that for F1 are thus constrained to be equal by having the same parameter name.

However, you must be careful not to provide too many parameters. For example, the following statement results in an error:

```
variance
  F1-F6 = 1.0 psi2-psi6 extra;
```

The parameters after psi6 are excessive.

### Default Variance Parameters

In the IRT model, by default, the factor variances are fixed at ones. You can also use the VARIANCE statement to override these default variance parameters in situations where you want to specify parameter constraints, provide initial or fixed values, or make parameter references.

---

## WEIGHT Statement

```
WEIGHT variable ;
```

The WEIGHT statement specifies the weight variable for the observations. The WEIGHT and FREQ statements have a similar effect, except that the WEIGHT statement does not alter the number of observations. An observation is used in the analysis only if the WEIGHT variable is greater than 0 and is not missing.

## Details: IRT Procedure

### Notation for the Item Response Theory Model

This section introduces the mathematical notation that is used throughout the chapter to describe the item response theory (IRT) model. For a description of the fitting algorithms and the mathematical-statistical details, see the section “Details: IRT Procedure” on page 4038.

A  $d$ -dimensional IRT model that has  $J$  ordinal responses can be expressed by the equations

$$y_{ij} = \lambda_j \eta_i + \epsilon_{ij}$$

$$p_{ijk} = \Pr(u_{ij} = k) = \Pr(\alpha_{(j,k-1)} < y_{ij} < \alpha_{(j,k)}), \quad k = 1, \dots, K$$

where  $u_{ij}$  is the observed ordinal response from subject  $i$  for item  $j$ ,  $y_{ij}$  is a continuous latent response that underlies  $u_{ij}$ ,  $\alpha_j = (\alpha_{(j,0)} = -\infty, \alpha_{(j,1)}, \dots, \alpha_{(j,K-1)}, \alpha_{(j,K)} = \infty)$  is a vector of threshold parameters for item  $j$ ,  $\lambda_j$  is a vector of slope (or discrimination) parameters for item  $j$ ,  $\eta_i = (\eta_{i1}, \dots, \eta_{id})$  is a vector of latent factors for subject  $i$ ,  $\eta_i \sim N_d(\mu, \Sigma)$ , and  $\epsilon_i = (\epsilon_{i1}, \dots, \epsilon_{iJ})$  is a vector of unique factors for subject  $i$ . All the unique factors in  $\epsilon_i$  are independent from one another, suggesting that  $y_{ij}$ ,  $j = 1, \dots, J$ , are independent conditional on the latent factor  $\eta_i$ . This is the so-called local independence assumption. Finally,  $\eta_i$  and  $\epsilon_i$  are also independent.

Based on the preceding model specification,

$$p_{ijk} = \int_{\alpha_{(j,k-1)} - \lambda_j \eta_i}^{\alpha_{(j,k)} - \lambda_j \eta_i} p(y; \lambda_j \eta_i, 1) dy = \int_{\alpha_{(j,k-1)} - \lambda_j \eta_i}^{\alpha_{(j,k)} - \lambda_j \eta_i} p(y; 0, 1) dy$$

where  $p$  is determined by the link function. It is the density function of the standard normal distribution if the probit link is used, or the density function of the logistic distribution if the logistic link is used.

Let  $\Lambda = (\lambda_1^T, \dots, \lambda_J^T)$  denote the slope matrix. To identify the model in exploratory analysis, the upper triangular elements of  $\Lambda$  are fixed as zero, the factor means  $\mu$  is fixed as a zero vector, and the factor variance covariance matrix  $\Sigma$  is fixed as an identity matrix. For confirmatory analysis, it is assumed that the identification problem is solved by user-specified constraints.

The model that is specified in the preceding equation uses the latent response formulation. PROC IRT uses this parameterization for computational convenience. When there is only one latent factor, a mathematically equivalent parameterization for the model is

$$p_{ijk} = \int_{-a_j(\eta_i - b_{j,k-1})}^{-a_j(\eta_i - b_{j,k})} p(y; 0, 1) dy$$

where  $a_j$  is called the slope (discrimination) parameter and  $b_{j,k}$ ,  $k = 1, \dots, K$  are called the threshold parameters. The threshold parameters under these two parameterizations can be translated as  $b_{j,k} = \frac{\alpha_{j,k}}{\lambda_j}$ , where  $k = 1, \dots, K$  and  $c_{j,k} = -\alpha_{j,k}$  is often called the intercept parameter.

The preceding model is called a graded-response model. When the responses are binary, this model reduces to the two-parameter model, which can be expressed as

$$y_{ij} = a_j(\eta_i - b_j) + \epsilon_{ij}$$

$$p_{ij} = \Pr(u_{ij} = 1) = \Pr(y_{ij} > 0)$$

where  $b_j$  is often called item difficulty parameter.

The two-parameter model reduces to a one-parameter model when slope parameters for all the items are constrained to be equal. In the case where logistic link is used, the one- and two-parameter models are often abbreviated as 1PL and 2PL. When all the slope parameters are set to 1 and the factor variance is set to a free parameter, the Rasch model is obtained.

You can obtain three- and four-parameter models by introducing the guessing and ceiling parameters. Let  $g_j$  and  $c_j$  denote the item-specific guessing and ceiling parameters. Then the four-parameter model can be expressed as

$$p_{ij} = \Pr(u_{ij} = 1) = g_j + (c_j - g_j) \Pr(y_{ij} > 0)$$

This model reduces to the three-parameter model when  $c_j = 1$ .

---

## Assumptions

The primary statistical assumptions that underlie the analyses that PROC IRT performs are as follows:

- The number of latent factors is known.
- Latent factors are assumed to be normally distributed.
- Conditional on latent factors, observed responses (items) are assumed to be independent. This is the so called local independence assumption.

These assumptions are necessary if you want to make statistical inferences. For exploratory analysis, these assumptions do not apply.

---

## PROC IRT Contrasted with Other SAS Procedures

IRT models are often referred to as latent trait models, especially in the field of sociology. The term latent trait is used to emphasize that observed discrete responses are manifestations of hypothesized traits, constructs, or attributes that cannot be directly observed. For that reason, IRT models belong to the more general modeling framework called latent variable models. Other models that belong to the latent variable model framework include factor analysis models, finite mixture models, and mixed effect models. The relationships between these different latent variable models can be described as shown in [Table 53.2](#).

**Table 53.2** Latent Variable Models

		Latent Variable	
		Continuous	Discrete
Observed Variable	Continuous	Factor analysis	Finite mixture model
	Discrete	Item response theory	Latent class analysis

This table suggests that latent variable models can be classified into four groups, based on the measurement scale of observed and latent variables. These different latent variable models can be fitted by different SAS procedures: PROC FACTOR for factor analysis models, PROC FMM for finite mixture models, and PROC IRT for item response theory models. IRT models are more closely related to factor analysis models. They can be considered a version of factor analysis models of discrete rather than continuous responses.

## Response Models

PROC IRT supports several response models for binary and ordinal responses, and it allows different items to have different response models. Details about these response models and their relationships follow:

- **One-parameter model:** This model assumes that items are binary. The distinctive feature of the one-parameter model, compared with the two-parameter model, is that the slopes (or the discrimination parameters) of the items are the same in the model. Statistically, the one-parameter model is equivalent to the Rasch model. They give the same model fit for the same data set.
- **Two-parameter model:** This model assumes that items are binary. The slopes (or the discrimination parameters) and the difficulty (or the intercept parameters) of the items are free parameters in the model. If all slopes of the two-parameter model are constrained to be same, it reduces to the one-parameter model.
- **Three-parameter model:** This model assumes that items are binary. The slopes (or the discrimination parameters), the difficulty (or the intercept parameters), and the guessing parameters of the items are free parameters in the model. If all the guessing parameters are fixed to 0, the three-parameter model reduces to the two-parameter model.
- **Four-parameter model:** This model assumes that items are binary. The slopes (or the discrimination parameters), the difficulty (or the intercept parameters), the guessing parameters, and the ceiling parameters of the items are free parameters in the model. If all the guessing parameters are fixed to 0 and all the ceiling parameters are fixed to 1, the four-parameter model reduces to the two-parameter model.
- **Rasch model:** This model assumes that items are binary. The distinctive feature of the Rasch model, compared with the two-parameter model, is that the slopes (or the discrimination parameters) of the items are all fixed to 1 (and with free factor variance parameters) in the model. Statistically, the Rasch model is equivalent to the one-parameter model. They give the same model fit for the same data set.
- **Graded response model:** This model assumes that items are ordinal-categorical with at most 19 levels. The slopes (or discrimination) the thresholds of the items are free parameters in the model.

You can specify the response function or model for all the variables that are listed in the **VAR** statement by using the **RESFUNC=** option in the **PROC IRT** statement. To specify different response functions or models for different set of variables, you can use the **MODEL** statement.

## Marginal Likelihood

Based on the model that is specified in the section “[Notation for the Item Response Theory Model](#)” on page 4038, the marginal likelihood is

$$L(\boldsymbol{\theta}|U) = \prod_{i=1}^N \int \prod_{j=1}^J \prod_{k=1}^K (P_{ijk})^{v_{ijk}} \phi(\boldsymbol{\eta}; \boldsymbol{\mu}, \boldsymbol{\Sigma}) d\boldsymbol{\eta} = \prod_{i=1}^N \int f(u_i|\boldsymbol{\eta}) \phi(\boldsymbol{\eta}; \boldsymbol{\mu}, \boldsymbol{\Sigma}) d\boldsymbol{\eta}$$

where  $v_{ijk} = I(u_{ij} = k)$ ,  $\phi(\boldsymbol{\eta})$  is the multivariate normal density function for the latent factor  $\boldsymbol{\eta}$ , and  $\boldsymbol{\theta}$  is a set of all the model parameters. The corresponding log likelihood is

$$\log L(\boldsymbol{\theta}|U) = \sum_{i=1}^N \log \int \prod_{j=1}^J \prod_{k=1}^K (P_{ijk})^{v_{ijk}} \phi(\boldsymbol{\eta}; \boldsymbol{\mu}, \boldsymbol{\Sigma}) d\boldsymbol{\eta}$$

Integrations in the preceding equation cannot be solved analytically and need to be approximated by using numerical integration,

$$\log \tilde{L}(\boldsymbol{\theta}|U) = \sum_{i=1}^N \log \left[ \sum_{g=1}^{G^d} \left[ \prod_{j=1}^J \prod_{k=1}^K (P_{ijk}(\mathbf{x}_g))^{v_{ijk}} \frac{\phi(\mathbf{x}_g; \boldsymbol{\mu}, \boldsymbol{\Sigma})}{\phi(\mathbf{x}_g; 0, I)} \right] w_g \right]$$

where  $d$  is the number of factors,  $G$  is the number of quadrature points per dimension, and  $\mathbf{x}_g$  and  $w_g$  are the quadrature points and weights, respectively.

## Approximating the Marginal Likelihood

As discussed in the section “[Marginal Likelihood](#)” on page 4041, integrations that are involved in the marginal likelihood for IRT model cannot be solved analytically and need to be approximated by using numerical integration, mostly Gauss-Hermite quadrature.

### Gauss-Hermite (G-H) Quadrature

In general, the Gauss-Hermite (G-H) quadrature can be presented as

$$\int_{-\infty}^{\infty} g(x) dx = \int_{-\infty}^{\infty} f(x) \phi(x) dx \approx \sum_{g=1}^G f(x_g) w_g$$

where  $G$  is the number of quadrature points and  $x_g$  and  $w_g$  are the integration points and weights, respectively, which are uniquely determined by the integration domain and the weighting kernel  $\phi(x)$ . Traditional G-H quadrature often uses  $e^{-x^2}$  as the weighting kernel. In the field of statistics, the density of standard normal distribution is more widely used instead, because for estimating various statistical models, the Gaussian density is often a factor of the integrand. In the case in which the Gaussian density is not a factor of the integrand, the integral is transformed into the form by dividing and multiplying the original integrand by the standard normal density.

## Adaptive Gauss-Hermite Quadrature

The G order G-H quadrature is exact if  $f(x)$  is a  $2K - 1$  degree polynomial in  $x$ . However, as many researchers (Lesaffre and Spiessens 2001; Rabe-Hesketh, Skrondal, and Pickles 2002) point out, integrands  $f(u_i|\eta)\phi(\eta; \mu, \Sigma)$  often have sharp peaks and cannot be well approximated by low-degree polynomials in  $\eta$ . Furthermore, the peak might be far from zero or be located between adjacent quadrature points so that substantial contributions to the integral are lost.

Note that the integrands in the marginal likelihood are a product of the prior density of  $\eta$ ,  $\phi(\eta; \mu, \Sigma)$  and the joint probability of responses given  $\eta$ ,  $f(u_i|\eta)$ . After normalization with respect to  $\eta$ , the integrand,  $f(u_i|\eta)\phi(\eta; \mu, \Sigma)$ , is just the posterior density of  $\eta$ , given the observed responses  $u_i$ . This posterior density is approximately normal when the number of items is large. Let  $\mu_i$  and  $\Sigma_i$  be the mean and covariance matrix, respectively, of the posterior density. Then the ratio  $\frac{f(u_i|\eta)\phi(\eta; \mu, \Sigma)}{\phi(\eta; \mu_i, \Sigma_i)}$  can be well approximated by a low-degree polynomial if the number of items is relatively large. This suggests that the integral should be transformed as

$$\int f(u_i|\eta)\phi(\eta)d\eta = \int \frac{f(u_i|\eta)\phi(\eta; \mu, \Sigma)}{\phi(\eta; \mu_i, \Sigma_i)}\phi(\eta; \mu_i, \Sigma_i)d\eta$$

The integration points and weights that correspond to  $\phi(\eta; \mu_i, \Sigma_i)$  are

$$z_g = \Sigma_i^{1/2}x_g + \mu_i$$

$$v_g = |\Sigma_i|^{1/2}w_g$$

The preceding transformations move and scale the quadrature points to the center of the integrands such that the integrand can be better approximated using many fewer quadrature points.

---

## Maximizing the Marginal Likelihood

You can obtain parameter estimates by maximizing the marginal likelihood by using either the expectation maximization (EM) algorithm or a Newton-type algorithm. Both algorithms are available in PROC IRT.

The most widely used estimation method for IRT models is the Gauss-Hermite quadrature-based EM algorithm, proposed by Bock and Aitkin (1981). However, this method has several important shortcomings, the most serious of which is the lack of reliable convergence criteria. Without reliable convergence criteria, estimates can be seriously biased because of spurious convergence. In comparison, gradient-based convergence criteria is readily available for Newton-type algorithms. As a result, PROC IRT uses the quasi-Newton algorithm instead of EM as the default optimization method.

## Newton-Type Algorithms

Newton-type algorithms maximize the marginal likelihood directly, based on the first and second derivatives. Two of the most widely used estimation algorithms are the Newton-Raphson and Fisher scoring algorithms, which rely on the gradient and Hessian of the log likelihood. However, for latent variable models that contain categorical responses, the Hessian matrix is often expensive to compute. As a result, several quasi-Newton algorithms that require only gradients have been proposed. In the field of IRT, Bock and Lieberman (1970) propose replacing the Hessian with the following information matrix:

$$I(\theta) = E \left[ \frac{\partial \log \tilde{L}(\theta|U)}{\partial \theta} \left( \frac{\partial \log \tilde{L}(\theta|U)}{\partial \theta} \right)^T \right] = \sum_{h=1}^{2^J} \left[ \frac{\partial \log \tilde{L}_i}{\partial \theta} \left( \frac{\partial \log \tilde{L}_i}{\partial \theta} \right)^T \right]$$



To calculate the preceding expectation, you need to sum over not just the observed but all  $2^J$  possible response patterns; this becomes computationally very expensive when the number of items is large. Fortunately, other quasi-Newton algorithms that do not have these computational difficulties have been proposed. The Broyden-Fletcher-Goldfarb-Shanno (BFGS) method is one of the most popular quasi-Newton algorithms that approximate the Hessian matrix with gradient.

For the objective function,  $\log \tilde{L}(\theta)$ , the first derivatives with respect to  $\theta_j$  are

$$\frac{\partial \log \tilde{L}(\theta | \mathbf{U})}{\partial \theta_j} = \sum_{i=1}^N \left[ (\tilde{L}_i)^{-1} \frac{\partial \tilde{L}_i}{\partial \theta_j} \right] = \sum_{i=1}^N \left[ (\tilde{L}_i)^{-1} \sum_{g=1}^{G^d} \left[ \frac{\partial f_i(\mathbf{x}_g)}{\partial \theta_j} w_g^* \right] \right]$$

where

$$\tilde{L}_i = \sum_{g=1}^{G^d} \left[ \prod_{j=1}^J \prod_{k=1}^K (P_{ijk}(\mathbf{x}_g))^{v_{ijk}} \frac{\phi(\mathbf{x}_g; \boldsymbol{\mu}, \boldsymbol{\Sigma})}{d(\mathbf{x}_g; 0, I)} \right] w_g = \sum_{g=1}^{G^d} f_i(\mathbf{x}_g) w_g$$

$$f_i(\mathbf{x}_g) = \prod_{j=1}^J \prod_{k=1}^K (P_{ijk}(\mathbf{x}_g))^{v_{ijk}} \frac{\phi(\mathbf{x}_g; \boldsymbol{\mu}, \boldsymbol{\Sigma})}{d(\mathbf{x}_g; 0, I)}$$

and

$$\frac{\partial f_i(\mathbf{x}_g)}{\partial \theta_{Ij}} = \frac{\partial [(P_{ijk}(\mathbf{x}_g))]}{\partial \theta_j} \frac{f_i(\mathbf{x}_g)}{(P_{ijk}(\mathbf{x}_g))}$$

$$\frac{\partial f_i(\mathbf{x}_g)}{\partial \theta_f} = \prod_{j=1}^J \prod_{k=1}^K (P_{ijk}(\mathbf{x}_g))^{v_{ijk}} \frac{\partial \phi(\mathbf{x}_g; \boldsymbol{\mu}, \boldsymbol{\Sigma})}{\partial \theta_f}$$

where, in the preceding two equations,  $\theta_{Ij}$  indicate parameters that are associated with item  $j$  and  $\theta_f$  represents parameters that are related to latent factors.

### Expectation-Maximization (EM) Algorithm

The expectation-maximization (EM) algorithm starts from the complete data log likelihood that can be expressed as follows:

$$\begin{aligned} \log L(\boldsymbol{\theta} | \mathbf{U}, \boldsymbol{\eta}) &= \sum_{i=1}^N \left[ \sum_{j=1}^J \sum_{k=1}^K v_{ijk} \log P_{ijk} + \log \phi(\boldsymbol{\eta}_i; \boldsymbol{\mu}, \boldsymbol{\Sigma}) \right] \\ &= \sum_{j=1}^J \sum_{i=1}^N \left[ \sum_{k=1}^K v_{ijk} \log P_{ijk} \right] + \sum_{i=1}^N \log \phi(\boldsymbol{\eta}_i; \boldsymbol{\mu}, \boldsymbol{\Sigma}) \end{aligned}$$

The expectation (E) step calculates the expectation of the complete data log likelihood with respect to the conditional distribution of  $\boldsymbol{\eta}_i$ ,  $f(\boldsymbol{\eta}_i | \mathbf{u}_i, \boldsymbol{\theta}^{(t)})$  as follows:

$$Q(\boldsymbol{\theta} | \boldsymbol{\theta}^{(t)}) = \sum_{j=1}^J \sum_{i=1}^N \left[ \sum_{k=1}^K v_{ijk} E \left[ \log P_{ijk} | \mathbf{u}_i, \boldsymbol{\theta}^{(t)} \right] \right] + \sum_{i=1}^N E \left[ \log \phi(\boldsymbol{\eta}_i; \boldsymbol{\mu}, \boldsymbol{\Sigma}) | \mathbf{u}_i, \boldsymbol{\theta}^{(t)} \right]$$

The conditional distribution  $f(\eta|u_i, \theta^{(t)})$  is

$$f(\eta|u_i, \theta^{(t)}) = \frac{f(u_i|\eta, \theta^{(t)})\phi(\eta; \mu^{(t)}, \Sigma^{(t)})}{\int f(u_i|\eta, \theta^{(t)})\phi(\eta; \mu^{(t)}, \Sigma^{(t)})d\eta} = \frac{f(u_i|\eta, \theta^{(t)})\phi(\eta; \mu^{(t)}, \Sigma^{(t)})}{f(u_i)}$$

These conditional expectations that are involved in the  $Q$  function can be expressed as follows:

$$E[\log P_{ijk}|u_i, \theta^{(t)}] = \int \log P_{ijk} f(\eta|u_i, \theta^{(t)})d\eta$$

$$E[\log \phi(\eta; \mu, \Sigma)|u_i, \theta^{(t)}] = \int \log \phi(\eta; \mu, \Sigma) f(\eta|u_i, \theta^{(t)})d\eta$$

Then

$$Q(\theta|\theta^{(t)}) = \sum_{j=1}^J \int [\log P_{ijk} r_{jk}(\theta^{(t)})] \phi(\eta; \mu^{(t)}, \Sigma^{(t)})d\eta + \int \log \phi(\eta|\theta) N(\theta^{(t)}) \phi(\eta; \mu^{(t)}, \Sigma^{(t)})d\eta$$

where

$$r_{jk}(\theta^{(t)}) = \sum_{i=1}^N \sum_{k=1}^K v_{ijk} \frac{f(u_i|\eta, \theta^{(t)})}{f(u_i)}$$

and

$$N(\theta^{(t)}) = \sum_{i=1}^N \frac{f(u_i|\eta, \theta^{(t)})}{f(u_i)}$$

Integrations in the preceding equations can be approximated as follows by using G-H quadrature:

$$\tilde{Q}(\theta|\theta^{(t)}) = \sum_{g=1}^G \left[ \log P_{ijk}(\mathbf{x}_g) r_{jk}(\mathbf{x}_g, \theta^{(t)}) + \log \phi(\mathbf{x}_g|\theta) N(\mathbf{x}_g, \theta^{(t)}) \right] \frac{\phi(\mathbf{x}_g; \mu^{(t)}, \Sigma^{(t)})}{d(\mathbf{x}_g; 0, I)} w_g$$

In the maximization (M) step of the EM algorithm, parameters are updated by maximizing  $\tilde{Q}(\theta|\theta^{(t)})$ . To summarize, the EM algorithm consists of the following two steps:

E step: Approximate  $Q(\theta|\theta^{(t)})$  by using numerical integration.

M step: Update parameter estimates by maximizing  $\tilde{Q}(\theta|\theta^{(t)})$  with the one-step Newton-Raphson algorithm.

---

## Factor Score Estimation

PROC IRT provides three methods of estimating factor scores: maximum likelihood (ML), maximum a posteriori (MAP), and expected a posteriori (EAP). You can specify them by using the **SCOREMETHOD=** option in the **PROC IRT** statement.

You can obtain the ML factor score by maximizing the likelihood for each observation with respect to the latent factor. You can also compute the MAP or EAP factor score by maximizing or by taking the expectation of the posterior distribution of latent factors for each observation. The likelihood and posterior distribution for each observation,  $u_i = (u_{i1}, \dots, u_{iJ})$ , can be expressed, respectively, as

$$l(\eta|u_i, \hat{\theta}) = \prod_{j=1}^J \prod_{k=1}^K (P_{ijk})^{v_{ijk}}$$

and

$$p(\eta|u_i, \hat{\theta}) \propto \prod_{j=1}^J \prod_{k=1}^K (P_{ijk})^{v_{ijk}} \phi(\eta; \mu, \Sigma)$$

Factor scores are restricted to the range from  $-99$  to  $99$ . For unidimensional models, the ML factor score is not available for subjects whose response to all the items is either the lowest or the highest level. For example, suppose there are five binary items in the model. For subjects whose response is 1 or 0 to all five items, the ML factor score cannot be estimated. For subjects whose response to all items is the lowest level, the ML factor score is set to  $-99$ , and for subjects whose response to all items is the highest level, the ML factor score is set to  $99$ .

---

## Model and Item Fit

The IRT procedure includes five model fit statistics: log likelihood, Akaike's information criterion (AIC), Bayesian information criterion (BIC), likelihood ratio  $G^2$ , and Pearson's chi-square.

The following two equations compute the likelihood ratio chi-square  $G^2$  and Pearson's chi-square,

$$G^2 = 2 \left[ \sum_{l=1}^L r_l \log \frac{r_l}{NP_l} \right]$$

$$\chi^2 = \sum_{l=1}^L \frac{(r_l - NP_l)^2}{NP_l}$$

where  $N$  is the number of subjects,  $L$  is number of possible response patterns,  $P_l$  is the estimated probability of observing response pattern  $l$ , and  $r_l$  is the number of subjects who have response pattern  $l$ . If the model is true, these two statistics asymptotically follow central chi-square distribution with degrees of freedom  $L - m - 1$ , where  $m$  is the number of free parameters in the model. When  $L$  (the number of possible response patterns) is much greater than  $N$ , the frequency table is sparse. This invalidates the use of chi-square distribution as the asymptotic distribution for these two statistics, and as a result the likelihood ratio chi-square and Pearson's chi-square statistics should not be used to evaluate overall model fit.

For item fit, PROC IRT computes the likelihood ratio  $G^2$  and Pearson's chi-square. Pearson's chi-square statistic, proposed by Yen (1981), has the form

$$Q_{1j} = \sum_{k=1}^{10} N_k \frac{(O_{jk} - E_{jk})^2}{E_{jk}(1 - E_{jk})}$$

The likelihood ratio  $G^2$ , proposed by McKinley and Mills (1985), uses the following equation:

$$G^2 = 2 \sum_{k=1}^{10} N_k \left[ O_{jk} \log \frac{O_{jk}}{E_{jk}} + (1 - O_{jk}) \log \frac{1 - O_{jk}}{1 - E_{jk}} \right]$$

These two statistics approximately follow a central chi-square distribution with  $10 - m_j$  degrees of freedom, where  $m_j$  is the number of free parameters for item  $j$ .

To calculate these two statistics, first order all the subjects according to their estimated factor scores, and then partition them into 10 intervals such that the number of subjects in each interval is approximately equal.  $O_{jk}$  and  $E_{jk}$  are the observed proportion and expected proportion, respectively, of subjects in interval  $k$  who have a correct response on item  $j$ . The expected proportions  $E_{jk}$  are computed as the mean predicted probability of a correct response in interval  $k$ .

---

## Item and Test information

Let  $P_{jk}(\theta)$  be the probability of endorsing category  $k$  for item  $j$  for a subject whose ability score is  $\theta$ . Then the item information function can be defined as

$$I_j(\theta) = \sum_{k=1}^K I_k(\theta) P_{jk}(\theta)$$

where

$$I_k(\theta) = -\frac{\partial^2}{\partial \theta^2} \log P_{jk}(\theta)$$

The test information function is the sum of the information functions of the items in the test. The information function of a test that has  $J$  items is

$$I(\theta) = \sum_{j=1}^J I_j(\theta)$$

---

## ODS Table Names

PROC IRT assigns a name to each table that it creates. You can use these names to refer to the table when you use the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in [Table 53.3](#). For more information about ODS, see Chapter 20, “Using the Output Delivery System.”

**Table 53.3** ODS Tables Produced by PROC IRT

ODS Table Name	Description	Option
ConvergenceStatus	Convergence Status	Default output

---

**Table 53.3** *continued*

ODS Table Name	Description	Option
Eigenvalues	Polychoric correlation–based eigenvalues	Default output
FactorCov	Factor covariance estimates	Default output; available only for multidimensional models
FactorCovInit	Initial factor covariance estimates	<b>PINITIAL</b> option; available only for multidimensional models
FactorCovRot	Rotated factor covariance estimates	<b>ROTATE=</b> oblique rotation methods; available only for multidimensional exploratory models
FitStatistics	Model fit statistics	Default output
GroupInfo	Group information	Default output; available only for multiple group analysis
ItemFit	Item fit statistics	<b>ITEMFIT</b> option; available only for binary responses that have one latent factor
ItemInfo	Item information	Default output
IterHistory	Iteration history	Default output
ModelInfo	Model information	Default output
OptInfo	Optimization information	Default output
ParameterEstimates	Item parameter (difficulty and slope) estimates for unidimensional models or intercept parameter estimates for multidimensional models	Default output
ParameterEstimatesInit	Initial item parameter (difficulty and slope) estimates for unidimensional model or initial intercept parameter estimates for multidimensional models	<b>PINITIAL</b> option
PolyCorr	Polychoric correlation matrix	<b>POLYCHORIC</b> option
Slope	Slope parameter estimates	Default output; available only for multidimensional confirmatory models
SlopeInit	Initial slope parameter estimates	<b>PINITIAL</b> option; available only for multidimensional models
SlopeRot	Rotated slope parameter estimates	Default output; available only for multidimensional exploratory models

## ODS Graphics

Statistical procedures use ODS Graphics to create graphs as part of their output. ODS Graphics is described in detail in Chapter 21, “[Statistical Graphics Using ODS](#).”

Before you create graphs, ODS Graphics must be enabled (for example, by specifying the ODS GRAPHICS ON statement). For more information about enabling and disabling ODS Graphics, see the section “[Enabling and Disabling ODS Graphics](#)” on page 606 in Chapter 21, “[Statistical Graphics Using ODS](#).”

The overall appearance of graphs is controlled by ODS styles. Styles and other aspects of using ODS Graphics are discussed in the section “[A Primer on ODS Statistical Graphics](#)” on page 605 in Chapter 21, “[Statistical Graphics Using ODS](#).”

You must also specify the **PLOTS=** option in the PROC IRT statement.

PROC IRT assigns a name to each graph that it creates using ODS. You can use these names to refer to the graphs when using ODS. The names are listed in [Table 53.4](#).

**Table 53.4** Graphs Produced by PROC IRT

ODS Graph Name	Plot Description	PLOTS= Option
ItemCharCurve	Item characteristic curves	<b>PLOTS=ICC</b> . The plot panels by default; specify <b>PLOTS=UNPACK</b> to produce an individual plot for each item.
ItemInfoCurve	Item information curves	<b>PLOTS=IIC</b> . The plot panels by default; specify <b>PLOTS=UNPACK</b> to produce an individual plot for each item.
PolyCorrHeatMap	Heat map for polychoric correlation matrix	<b>PLOTS=POLYCHORIC PLCORR</b>
ScreePlot	Scree and variance explained plots	<b>PLOTS=SCREE</b>
TestInfoCurve	Test information curve	<b>PLOTS=TIC</b>
VariancePlot	Plot of explained variance	<b>PLOTS=SCREE(UNPACK)</b>

## Examples: IRT Procedure

### Example 53.1: Unidimensional IRT Models

This example shows you the features that PROC IRT provides for unidimensional analysis. The data set comes from the 1978 Quality of American Life Survey. The survey was administered to a sample of all U.S. residents aged 18 years and older in 1978. In this survey, subjects were asked to rate their satisfaction with

many different aspects of their lives. This example selects eight items. These items are designed to measure people's satisfaction in the following areas on a seven-point scale: community, neighborhood, dwelling unit, life in the United States, amount of education received, own health, job, and how spare time is spent. For illustration purposes, the first five items are dichotomized and the last three items are collapsed into three levels.

The following DATA step creates the data set IrtUni.

```
data IrtUni;
    input item1-item8 @@;
    datalines;
1 0 0 0 1 1 2 1 1 1 1 1 1 3 3 3 0 1 0 0 1 1 1 1 1 0 0 1 0 1 2 3 0 0 0
0 0 1 1 1 1 0 0 1 0 1 3 3 0 0 0 0 0 1 1 3 0 0 1 0 0 1 2 2 0 1 0 0 1 1

    ... more lines ...

3 3 0 1 0 0 1 2 2 1
;
```

Because all the items are designed to measure subjects' satisfaction in different aspects of their lives, it is reasonable to start with a unidimensional IRT model. The following statements fit such a model by using several user-specified options:

```
ods graphics on;
proc irt data=IrtUni link=probit pinitial itemfit plots=ICC;
    var item1-item8;
    model item1-item4/resfunc=twop, item5-item8/resfunc=graded;
run;
ods graphics off;
```

The ODS GRAPHICS ON statement invokes the ODS Graphics environment and displays the plots, such as the item characteristic curve plot. For more information about ODS Graphics, see Chapter 21, “[Statistical Graphics Using ODS](#).”

The first option is the [LINK=](#) option, which specifies that the link function be the probit link. Next, you request initial parameter estimates by using the [PINITIAL](#) option. Item fit statistics are displayed using the [ITEMFIT](#) option. In the [PROC IRT](#) statement, you can use the [PLOTS](#) option to request different plots. In this example, you request item characteristic curves by using the [PLOTS=ICC](#) option.

In this example, you use the [MODEL](#) statement to specify different response models for different items. The specifications in the [MODEL](#) statement suggest that the first four items, item1 to item4, are fitted using the two-parameter model, whereas the last four items, item5 to item8, are fitted using the graded response model.

[Output 53.1.1](#) displays two tables. From the “Modeling Information” table, you can observe that the link function has changed from the default LOGIT link to the specified PROBIT link. The “Item Information” table shows that item1 to item5 each have two levels and item6 to item8 each have three levels. The last column shows the raw values of these different levels.

**Output 53.1.1** Basic Information**The IRT Procedure**

Modeling Information	
Data Set	WORK.IRTUNI
Link Function	Probit
Number of Items	8
Number of Factors	1
Number of Observations Read	500
Number of Observations Used	500
Estimation Method	Marginal Maximum Likelihood

Item Information			
Response Model	Item	Levels	Values
TwoP	item1	2	0 1
	item2	2	0 1
	item3	2	0 1
	item4	2	0 1
Graded	item5	2	0 1
	item6	3	1 2 3
	item7	3	1 2 3
	item8	3	1 2 3

PROC IRT produces the “Eigenvalues of the Polychoric Correlation Matrix” table in [Output 53.1.2](#) by default. You can use these eigenvalues to assess the dimension of latent factors. For this example, the fact that only the first eigenvalue is greater than 1 suggests that a one-factor model for the items is reasonable.

**Output 53.1.2** Eigenvalues of Polychoric Correlations**The IRT Procedure**

Eigenvalues of the Polychoric Correlation Matrix				
	Eigenvalue	Difference	Proportion	Cumulative
1	3.11870486	2.12497677	0.3898	0.3898
2	0.99372809	0.10025986	0.1242	0.5141
3	0.89346823	0.03116998	0.1117	0.6257
4	0.86229826	0.10670185	0.1078	0.7335
5	0.75559640	0.17795713	0.0944	0.8280
6	0.57763928	0.10080017	0.0722	0.9002
7	0.47683911	0.15511333	0.0596	0.9598
8	0.32172578		0.0402	1.0000

The **PINITIAL** option in the **PROC IRT** statement displays the “Initial Item Parameter Estimates” table, shown in [Output 53.1.3](#).



**Output 53.1.3** Initial Parameter Estimates**The IRT Procedure**

Initial Item Parameter Estimates			
Response Model	Item	Parameter	Estimate
TwoP	item1	Difficulty	0.26428
		Slope	1.05346
	item2	Difficulty	0.58640
		Slope	0.93973
	item3	Difficulty	0.44607
		Slope	0.82826
	item4	Difficulty	0.50157
		Slope	0.50906
Graded	item5	Threshold 1	-0.86792
		Slope	0.41380
	item6	Threshold 1	-0.59512
		Threshold 2	2.00678
		Slope	0.36063
	item7	Threshold 1	-0.90743
		Threshold 2	0.69335
		Slope	0.64191
	item8	Threshold 1	-1.18209
		Threshold 2	0.26959
		Slope	0.67591

Output 53.1.4 includes tables that are related to the optimization. The “Optimization Information” table shows that the log likelihood is approximated by using seven adaptive Gauss-Hermite quadrature points and then maximized by using the quasi-Newton algorithm. The number of free parameters in this example is 19. The “Iteration History” table shows the number of function evaluations, the objective function (–log likelihood divided by number of subjects) values, the objective function change, and the maximum gradient for each iteration. This information is very useful in monitoring the optimization status. Output 53.1.4 shows the convergence status at the bottom. The optimization converges according to the GCONV=0.00000001 criterion.

**Output 53.1.4** Optimization Information**The IRT Procedure**

Optimization Information	
Optimization Technique	Quasi-Newton
Likelihood Approximation	Adaptive Gauss-Hermite Quadrature
Number of Quadrature Points	9
Number of Free Parameters	19

**Output 53.1.4** *continued*

Iteration History				
Iteration	Evaluations	Objective		Max
		Function	Change	Gradient
0	2	6.19423730	6.19423730	0.015501
1	5	6.19269781	-0.00153950	0.005787
2	8	6.19256587	-0.00013194	0.003813
3	10	6.19249856	-0.00006732	0.003278
4	12	6.19245371	-0.00004485	0.004635
5	15	6.19243639	-0.00001732	0.001284
6	18	6.19242942	-0.00000696	0.00049
7	21	6.19242884	-0.00000058	0.000191
8	24	6.19242871	-0.00000013	0.000104
9	27	6.19242867	-0.00000004	0.000051
10	30	6.19242867	-0.00000001	0.000011

Convergence criterion (GCONV=.000000010) satisfied.

Output 53.1.5 displays the model fit and item fit statistics. Note that the item fit statistics apply only to the binary items. That is why these fit statistics are missing for item6 to item8.

**Output 53.1.5** Fit Statistics**The IRT Procedure**

Model Fit Statistics							
Log Likelihood				-3096.214333			
AIC (Smaller is Better)				6230.4286657			
BIC (Smaller is Better)				6310.5062196			

Item Fit Statistics							
Response Model	Item	DF	Pearson		LR		
			Chi-Square	Pr > P	ChiSq	Chi-Square	Pr > LR
TwoP	item1	8	34.16545	<.0001		49.39743	<.0001
	item2	8	30.34646	0.0002		37.52865	<.0001
	item3	8	27.54546	0.0006		36.34505	<.0001
	item4	8	22.75981	0.0037		26.13409	0.0010
Graded	item5	8	18.32369	0.0189		19.68488	0.0116
	item6	.	.	.	.	.	.
	item7	.	.	.	.	.	.
	item8	.	.	.	.	.	.

The last table for this example is the “Item Parameter Estimates ” table in [Output 53.1.6](#). This table contains parameter estimates, standard errors, and  $p$ -values. These  $p$ -values suggest that all the parameters are significantly different from zero.


### Output 53.1.6 Parameter Estimates

#### The IRT Procedure


Item Parameter Estimates					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
TwoP	item1	Difficulty	0.27342	0.08301	0.0005
		Slope	0.98381	0.14145	<.0001
	item2	Difficulty	0.60268	0.10047	<.0001
		Slope	0.90009	0.13112	<.0001
	item3	Difficulty	0.46110	0.10061	<.0001
		Slope	0.79521	0.11392	<.0001
	item4	Difficulty	0.50686	0.14410	0.0002
		Slope	0.50431	0.08567	<.0001
Graded	item5	Threshold	-0.79750	0.18708	<.0001
		Slope	0.45385	0.08238	<.0001
	item6	Threshold 1	-0.59139	0.19858	0.0014
		Threshold 2	2.02708	0.39593	<.0001
		Slope	0.35770	0.06777	<.0001
	item7	Threshold 1	-0.82130	0.12753	<.0001
		Threshold 2	0.64441	0.11431	<.0001
		Slope	0.72674	0.09312	<.0001
	item8	Threshold 1	-1.08124	0.14132	<.0001
		Threshold 2	0.25166	0.09535	0.0042
		Slope	0.76383	0.09753	<.0001

Item characteristic curves (ICC) are also produced in this example. By default, these ICC plots are displayed in panels. To display an individual ICC plot for each item, use the UNPACK suboption in the **PLOTS=** option in the **PROC IRT** statement.

Output 53.1.7 ICC Plots



Output 53.1.7 continued



Now, suppose your research hypothesis includes some equality constraints on the model parameters—for example, the slopes for the first four items are equal. Such equality constraints can be specified easily by using the **EQUALITY** statement. In the following example, the slope parameters of the first four items are equal:

```
proc irt data=IrtUni;
  var item1-item8;
  model item1-item4/resfunc=twop, item5-item8/resfunc=graded;
  equality item1-item4/parm=[slope];
run;
```

To estimate the factor score for each subject and add these scores to the original data set, you can use the **OUT=** option in the **PROC IRT** statement. PROC IRT provides three factor score estimation methods: maximum likelihood (ML), maximum a posteriori (MAP), and expected a posteriori (EAP). You can choose an estimation method by using the **SCOREMETHOD=** option in the **PROC IRT** statement. The default method is maximum a posteriori. In the following, factor scores along with the original data are saved to a SAS data set called `IrtUniFscore`:

```
proc irt data=IrtUni out=IrtUniFscore;
  var item1-item8;
  model item1-item4/resfunc=twop,
        item5-item8/resfunc=graded;
  equality item1-item4/parm=[slope];
run;
```

Sometimes you might find it useful to sort the items based on the estimated difficulty or slope parameters. You can do this by outputting the ODS tables for the estimates into data sets and then sorting the items by using PROC SORT. A simulated data set is used to show the steps.

The following DATA step creates the data set IrtSimu:

```
data IrtSimu;
  input item1-item25 @@;
  datalines;
1 1 1 0 1 1 0 0 1 1 0 0 0 0 1 0 1 1 1 0 0 0 1 0 1 1 1 1 0 0
0 0 0 0 0 1 1 0 0 0 1 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
1 1 1 1 1 1 1 1 1 0 0 0 0 0 0 0 0 0 1 0 0 0 1 0 0 0 0 1 0 0
1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 0 0 1 0 0 0 1 1 0 0
0 1 1 1 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1 0 1 0 0 0 1 0 1 1 0 1
1 0 0 1 1 0 0 0 1 0 0 0 0 0 0 0 1 0 1 0 1 0 1 0 0 1 1 1 1 1
1 1 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 0 0 1 1
1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 0 1 0 1 1 1 1 1 1 1 0 1 1

... more lines ...

1 1 0 1 1 1 1 1 1 1 0 1 1 0 0 0 1 1 0 1 1 1 1 1 1 0 0 1 0 1
0 0 0 0 0 0 0 0 1 1 1 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1 0 1 1
1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1
1 1 1 1 1 1 1 1 0 0 1 1 1 0 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
;
```

First, you build the model and output the parameter estimates table into a SAS data set by using the ODS OUTPUT statement:

```
proc irt data=IrtSimu link=probit;
  var item1-item25;
  ods output ParameterEstimates=ParmEst;
run;
```

Output 53.1.8 shows the “Item Parameter Estimates” table. Notice that the difficulty and slope parameters are in the same column. The reason for this is to avoid having an extremely wide table when each item has a lot of parameters.

**Output 53.1.8** Basic Information**The IRT Procedure**

Item Parameter Estimates				
Item	Parameter	Estimate	Standard Error	Pr >  t
item1	Difficulty	-1.32543	0.09769	<.0001
	Slope	1.44263	0.16091	<.0001
item2	Difficulty	-0.99702	0.07431	<.0001
	Slope	1.82221	0.19006	<.0001
item3	Difficulty	-1.24965	0.08962	<.0001
	Slope	1.58762	0.17495	<.0001
item4	Difficulty	-1.09577	0.07727	<.0001
	Slope	1.86821	0.20452	<.0001
item5	Difficulty	-1.07857	0.07784	<.0001
	Slope	1.78392	0.19080	<.0001
item6	Difficulty	-0.95057	0.09377	<.0001
	Slope	1.04203	0.10284	<.0001
item7	Difficulty	-0.65085	0.06918	<.0001
	Slope	1.45413	0.13473	<.0001
item8	Difficulty	-0.76372	0.07583	<.0001
	Slope	1.30444	0.12230	<.0001
item9	Difficulty	-0.72281	0.07030	<.0001
	Slope	1.50740	0.14242	<.0001
item10	Difficulty	-0.50750	0.06091	<.0001
	Slope	1.82387	0.17155	<.0001
item11	Difficulty	-0.01360	0.06433	0.4163
	Slope	1.26234	0.11287	<.0001
item12	Difficulty	0.04013	0.05537	0.2343
	Slope	2.02257	0.20099	<.0001
item13	Difficulty	0.16034	0.06845	0.0096
	Slope	1.13120	0.10199	<.0001
item14	Difficulty	0.01069	0.05625	0.4246
	Slope	1.89082	0.18128	<.0001
item15	Difficulty	0.07152	0.07328	0.1645
	Slope	0.96388	0.09050	<.0001
item16	Difficulty	-0.81412	0.07905	<.0001
	Slope	1.25375	0.11885	<.0001
item17	Difficulty	-0.92046	0.09289	<.0001
	Slope	1.02930	0.10125	<.0001
item18	Difficulty	-0.59407	0.06606	<.0001
	Slope	1.58443	0.14868	<.0001
item19	Difficulty	-0.97598	0.09743	<.0001
	Slope	0.97979	0.09761	<.0001
item20	Difficulty	-0.48859	0.05960	<.0001
	Slope	1.95710	0.18827	<.0001
item21	Difficulty	-0.60655	0.06819	<.0001
	Slope	1.45324	0.13427	<.0001
item22	Difficulty	-0.51263	0.06188	<.0001
	Slope	1.74472	0.16251	<.0001
item23	Difficulty	-0.90928	0.08451	<.0001

**Output 53.1.8** *continued***The IRT Procedure**

Item Parameter Estimates				
Item	Parameter	Estimate	Standard Error	Pr >  t
	Slope	1.20281	0.11623	<.0001
item24	Difficulty	-0.56515	0.06295	<.0001
	Slope	1.74437	0.16384	<.0001
item25	Difficulty	-0.58905	0.06718	<.0001
	Slope	1.48957	0.13785	<.0001

**Output 53.1.9** The Difficulty Parameter SAS Data Set

Obs	Item	Difficulty
1	item1	-1.32543
2	item2	-0.99702
3	item3	-1.24965
4	item4	-1.09577
5	item5	-1.07857
6	item6	-0.95057
7	item7	-0.65085
8	item8	-0.76372
9	item9	-0.72281
10	item10	-0.50750
11	item11	-0.01360
12	item12	0.04013
13	item13	0.16034
14	item14	0.01069
15	item15	0.07152
16	item16	-0.81412
17	item17	-0.92046
18	item18	-0.59407
19	item19	-0.97598
20	item20	-0.48859
21	item21	-0.60655
22	item22	-0.51263
23	item23	-0.90928
24	item24	-0.56515
25	item25	-0.58905

Then you save the estimates of slopes and difficulties in the data set ParmEst and create two separate data sets to store the difficulty and slope parameters:

```
data Diffs(keep=Item Difficulty);
  set ParmEst;
  Difficulty = Estimate;
  if (Parameter = "Difficulty") then output;
run;
```



```

proc print data=Diffs;
run;

data Slopes(keep=Item Slope);
  set ParmEst;
  Slope = Estimate;
  if (Parameter = "Slope") then output;
run;
proc print data=Slopes;
run;

```

The two SAS data sets are shown in [Output 53.1.9](#) and [Output 53.1.10](#).

**Output 53.1.10** The Slope Parameter SAS Data Set

Obs	Item	Slope
1	item1	1.44263
2	item2	1.82221
3	item3	1.58762
4	item4	1.86821
5	item5	1.78392
6	item6	1.04203
7	item7	1.45413
8	item8	1.30444
9	item9	1.50740
10	item10	1.82387
11	item11	1.26234
12	item12	2.02257
13	item13	1.13120
14	item14	1.89082
15	item15	0.96388
16	item16	1.25375
17	item17	1.02930
18	item18	1.58443
19	item19	0.97979
20	item20	1.95710
21	item21	1.45324
22	item22	1.74472
23	item23	1.20281
24	item24	1.74437
25	item25	1.48957

Now you can use PROC SORT to sort the items by either difficulty or slope as follows:

```

proc sort data=Diffs;
  by Difficulty;
run;
proc print data=Diffs;
run;

```

```

proc sort data=Slopes;
  by Slope;
run;
proc print data=Slopes;
run;

```

Output 53.1.11 and Output 53.1.12 show the sorted data sets.

**Output 53.1.11** Items Sorted by Difficulty

Obs	Item	Difficulty
1	item1	-1.32543
2	item3	-1.24965
3	item4	-1.09577
4	item5	-1.07857
5	item2	-0.99702
6	item19	-0.97598
7	item6	-0.95057
8	item17	-0.92046
9	item23	-0.90928
10	item16	-0.81412
11	item8	-0.76372
12	item9	-0.72281
13	item7	-0.65085
14	item21	-0.60655
15	item18	-0.59407
16	item25	-0.58905
17	item24	-0.56515
18	item22	-0.51263
19	item10	-0.50750
20	item20	-0.48859
21	item11	-0.01360
22	item14	0.01069
23	item12	0.04013
24	item15	0.07152
25	item13	0.16034

**Output 53.1.12** Items Sorted by Slope

Obs	Item	Slope
1	item15	0.96388
2	item19	0.97979
3	item17	1.02930
4	item6	1.04203
5	item13	1.13120
6	item23	1.20281
7	item16	1.25375
8	item11	1.26234
9	item8	1.30444
10	item1	1.44263
11	item21	1.45324
12	item7	1.45413
13	item25	1.48957
14	item9	1.50740
15	item18	1.58443
16	item3	1.58762
17	item24	1.74437
18	item22	1.74472
19	item5	1.78392
20	item2	1.82221
21	item10	1.82387
22	item4	1.86821
23	item14	1.89082
24	item20	1.95710
25	item12	2.02257

Notice that the sorting does not work correctly if any of the items have more than one threshold (ordinal response) or slope (multidimensional model).

Now, suppose you want to group the items into subgroups based on their difficulty parameters and then sort the items in each subgroup by their slope parameters. First, you need to merge the two data sets, *Diff*s and *Slope*s, into one data set. Then, you add another variable, called *DiffLevel*, to indicate the subgroups. The following statements show these steps:

```
proc sort data=Slopes;
  by Item;
run;
proc sort data=Diff;
  by Item;
run;
data ItemEst;
  merge Diff Slopes;
  by Item;
  if Difficulty < -1.0 then DiffLevel = 1;
  else if Difficulty < 0 then DiffLevel = 2;
  else if Difficulty < 1 then DiffLevel = 3;
  else DiffLevel = 4;
run;
```

```
proc print data=ItemEst;
run;
```

Output 53.1.13 shows the merged data set.

**Output 53.1.13** The Merged SAS Data Set

Obs	Item	Difficulty	Slope	DiffLevel
1	item1	-1.32543	1.44263	1
2	item10	-0.50750	1.82387	2
3	item11	-0.01360	1.26234	2
4	item12	0.04013	2.02257	3
5	item13	0.16034	1.13120	3
6	item14	0.01069	1.89082	3
7	item15	0.07152	0.96388	3
8	item16	-0.81412	1.25375	2
9	item17	-0.92046	1.02930	2
10	item18	-0.59407	1.58443	2
11	item19	-0.97598	0.97979	2
12	item2	-0.99702	1.82221	2
13	item20	-0.48859	1.95710	2
14	item21	-0.60655	1.45324	2
15	item22	-0.51263	1.74472	2
16	item23	-0.90928	1.20281	2
17	item24	-0.56515	1.74437	2
18	item25	-0.58905	1.48957	2
19	item3	-1.24965	1.58762	1
20	item4	-1.09577	1.86821	1
21	item5	-1.07857	1.78392	1
22	item6	-0.95057	1.04203	2
23	item7	-0.65085	1.45413	2
24	item8	-0.76372	1.30444	2
25	item9	-0.72281	1.50740	2

Then, you can sort the items by slope within each difficulty group as follows:

```
proc sort data=ItemEst;
  by difflevel slope;
run;
proc print data=ItemEst;
run;
```

Output 53.1.14 shows the data set after sorting.

**Output 53.1.14** Item Sorted by Slope within Each Difficulty Group

Obs	Item	Difficulty	Slope	DiffLevel
1	item1	-1.32543	1.44263	1
2	item3	-1.24965	1.58762	1
3	item5	-1.07857	1.78392	1
4	item4	-1.09577	1.86821	1
5	item19	-0.97598	0.97979	2
6	item17	-0.92046	1.02930	2
7	item6	-0.95057	1.04203	2
8	item23	-0.90928	1.20281	2
9	item16	-0.81412	1.25375	2
10	item11	-0.01360	1.26234	2
11	item8	-0.76372	1.30444	2
12	item21	-0.60655	1.45324	2
13	item7	-0.65085	1.45413	2
14	item25	-0.58905	1.48957	2
15	item9	-0.72281	1.50740	2
16	item18	-0.59407	1.58443	2
17	item24	-0.56515	1.74437	2
18	item22	-0.51263	1.74472	2
19	item2	-0.99702	1.82221	2
20	item10	-0.50750	1.82387	2
21	item20	-0.48859	1.95710	2
22	item15	0.07152	0.96388	3
23	item13	0.16034	1.13120	3
24	item14	0.01069	1.89082	3
25	item12	0.04013	2.02257	3

**Example 53.2: Multidimensional Exploratory and Confirmatory IRT Models**

This example illustrates how to use the IRT procedure to fit multidimensional exploratory and confirmatory IRT models. The data set that is introduced in [Example 53.1](#) is also used here. Two more items, item9 and item10, are added to the data set. These two items are designed to measure subjects' satisfaction with their friendships and their family life, respectively.

```
data IrtMulti;
  input item1-item10 @@;
  datalines;
1 0 0 0 1 1 2 1 2 1 1 1 1 1 1 3 3 3 3 3 0 1 0 0 1 1 1 1 1 1 1 0 0 1 0
1 2 3 2 2 0 0 0 0 0 1 1 1 1 1 1 0 0 1 0 1 3 3 1 2 0 0 0 0 0 1 1 3 3 2
0 0 1 0 0 1 2 2 3 2 0 1 0 0 1 1 1 2 2 2 0 0 0 0 0 2 2 3 3 2 0 1 0 1 0
2 3 3 3 3 0 0 1 0 1 1 2 3 2 3 1 1 1 1 1 2 2 3 2 2 0 0 0 0 1 1 2 2 3 1
1 0 1 1 1 2 3 3 2 3 0 1 0 0 1 1 2 3 3 3 1 0 1 1 1 2 3 3 3 3 0 1 0 1 1
3 2 3 3 2 1 1 1 0 0 1 3 3 2 1 1 1 0 0 1 2 3 3 3 3 0 1 1 1 1 1 2 1 2 3
1 0 0 1 1 3 1 1 1 1 1 0 0 0 1 1 1 3 3 1 0 0 0 0 1 1 3 3 3 3 0 0 0 0 0
1 1 1 3 2 1 0 0 0 0 1 3 3 3 3 1 1 0 1 1 3 1 1 3 3 1 0 1 1 1 1 3 1 1 1
... more lines ...
```

```

3 3 1 3 2 0 0 0 1 0 1 3 2 2 1 0 0 0 0 1 1 2 2 2 3 1 0 1 0 1 2 2 3 2 1
1 0 0 1 1 1 2 2 3 1 0 1 0 1 0 1 3 1 1 1 0 1 1 0 1 3 3 3 3 2 1 0 1 0 0
1 2 1 1 1 1 0 1 1 0 1 3 3 1 3 1 1 0 1 0 2 2 2 2 3 1 1 0 1 1 3 2 3 2 2
0 0 0 1 0 2 2 3 1 2 0 0 0 1 0 2 3 3 3 2 0 1 0 0 1 2 2 1 2 1
;

```

Now, suppose that previous research results suggest that two latent factors underlie these 10 items. However, knowledge about the factor structure is very limited. The first step you can take is to fit an exploratory IRT model by using two factors. This can be accomplished easily by submitting the following statements:

```


ods graphics on;
proc irt data=IrtMulti nfactor=2 plots=scree;
    var item1-item10;
run;

```

The first table that you want to check is the “Eigenvalue” table, shown in [Output 53.2.1](#). There are only two eigenvalues greater than 1 in this example. This result, to some extent, suggests that two factors might be enough in this example. [Output 53.2.2](#) include the scree and variance explained plots.

**Output 53.2.1** Eigenvalues of the Polychoric Correlation matrix  
**The IRT Procedure**

Eigenvalues of the Polychoric Correlation Matrix				
	Eigenvalue	Difference	Proportion	Cumulative
1	3.65750431	2.47883117	0.3658	0.3658
2	1.17867314	0.23738137	0.1179	0.4836
3	0.94129177	0.04672399	0.0941	0.5777
4	0.89456778	0.07508308	0.0895	0.6672
5	0.81948471	0.14774320	0.0819	0.7492
6	0.67174151	0.12081203	0.0672	0.8163
7	0.55092948	0.01698300	0.0551	0.8714
8	0.53394648	0.08647230	0.0534	0.9248
9	0.44747418	0.14308753	0.0447	0.9696
10	0.30438664		0.0304	1.0000

**Output 53.2.2** Scree and Variance Explained Plots

If the optimization algorithm converges successfully, the original and rotated slope matrices are produced. The default rotation method is varimax. You can use the **ROTATE=** option in the **PROC IRT** statement to specify a different rotation method.

**Output 53.2.3** Slope Matrix

Rotated Slope Matrix		
	_Factor1	_Factor2
item1	1.87474	0.63566
item2	1.71878	0.56368
item3	1.29600	0.54156
item4	0.74920	0.36351
item5	0.41017	0.63813
item6	0.44038	0.39948
item7	0.68674	1.18496
item8	0.74058	1.97319
item9	0.40213	1.26228
item10	0.40067	1.26536

This example uses the default varimax rotation. [Output 53.2.3](#) shows the rotated slope matrices. The rotated slope matrix is displayed in the standard matrix format. From the rotated slope matrix, you can see that the first factor is mainly reflected by item1 to item4 and item6, and the second factor is mainly reflected by the rest of the items. The exploratory results suggest a hypothesis about the factor structure of the items. In practice, you might want to confirm this structure by a confirmatory analysis of the new data. However, for illustration purposes, the same data set is used here to demonstrate the confirmatory model fitting by using the following statements:

```
proc irt data=IrtMulti;
  var item1-item4 item6 item5 item7-item10;
  factor Factor1->item1-item4 item6,
         Factor2->item5 item7-item10;
run;
ods graphics off;
```

Output 53.2.5 and Output 53.2.4 show the model fit statistics for the confirmatory model and the exploratory model, respectively. Output 53.2.6 shows the slope matrix from this confirmatory analysis. Because the number of response patterns in the data set, 500, is much lower than the total number of possible response patterns,  $2^5 \times 3^5 = 7,776$ , you cannot use Pearson's chi-square or the likelihood ratio statistic to test the overall fit of the confirmatory model.

If you have two nested models, you can still use the likelihood ratio statistic to test the difference between these two models. For illustration purposes, assume that the exploratory model and the confirmatory model are two independent models. Because the confirmatory model is nested within the exploratory model, you can use the likelihood ratio test to compare the two models. The likelihood ratio test statistic is 132.4. The degree of freedom is 9. The corresponding  $p$ -value is 0, which suggests that the difference between these two models is significant. There are many options that you can use to improve the model fit. For this example, you can try to free some of the fixed slope parameters or free the covariance between the factors.

#### Output 53.2.4 Model Fit Statistics for Exploratory Model

##### The IRT Procedure

Model Fit Statistics	
Log Likelihood	-3993.389529
AIC (Smaller is Better)	8054.7790586
BIC (Smaller is Better)	8198.075734

#### Output 53.2.5 Model Fit Statistics for Confirmatory Model

##### The IRT Procedure

Model Fit Statistics	
Log Likelihood	-4125.792484
AIC (Smaller is Better)	8301.5849679
BIC (Smaller is Better)	8406.9501703



**Output 53.2.6** Slope Matrix

Slope Matrix		
Estimate/StdErr/p-value		
	Factor1	Factor2
item1	2.04932	0.00000
	0.41637	.
	0.00000	.
item2	1.77313	0.00000
	0.32285	.
	0.00000	.
item3	1.36097	0.00000
	0.22842	.
	0.00000	.
item4	0.80650	0.00000
	0.15641	.
	0.00000	.
item6	0.55226	0.00000
	0.12128	.
	0.00000	.
item5	0.00000	-0.75216
	.	0.14089
	.	0.00000
item7	0.00000	-1.36470
	.	0.17608
	.	0.00000
item8	0.00000	-2.29997
	.	0.37170
	.	0.00000
item9	0.00000	-1.33480
	.	0.17139
	.	0.00000
item10	0.00000	-1.34809
	.	0.17566
	.	0.00000

**Example 53.3: Multiple-Group Analysis**

This example shows how to use the IRT procedure to do multiple-group analysis. The following DATA step creates the data set IrtGroup:

```
data IrtGroup;
  input item1-item8 GroupVar @@;
  datalines;
1 0 0 0 1 1 2 1 2 1 1 1 1 1 1 3 3 3 1 0 1 0 0 1 1 1 1 1 0 0 1 0 1 2 3
2 0 0 0 0 0 1 1 1 1 1 0 0 1 0 1 3 3 1 0 0 0 0 0 1 1 3 2 0 0 1 0 0 1 2
2 1 0 1 0 0 1 1 1 2 1 0 0 0 0 0 2 2 3 1 0 1 0 1 0 2 3 3 2 0 0 1 0 1 1
2 3 1 1 1 1 1 1 2 2 3 2 0 0 0 0 1 1 2 2 1 1 0 1 1 1 2 3 3 2 0 1 0 0 1
1 2 3 2 1 0 1 1 1 2 3 3 2 0 1 0 1 1 3 2 3 2 1 1 1 0 0 1 3 3 1 1 1 0 0
1 2 3 3 2 0 1 1 1 1 1 2 1 2 1 0 0 1 1 3 1 1 2 1 0 0 0 1 1 1 3 2 0 0 0
0 1 1 3 3 1 0 0 0 0 0 1 1 1 2 1 0 0 0 0 1 3 3 2 1 1 0 1 1 3 1 1 1 1 0
1 1 1 1 3 1 2 1 1 1 1 1 2 3 2 2 0 0 1 0 0 2 2 2 1 0 0 1 0 1 1 2 3 2 1
... more lines ...
```

```

1 0 0 2 1 3 2 1 1 1 1 1 1 3 2 1 1 1 1 0 0 3 3 1 2 1 0 0 1 1 3 3 3 2 0
0 1 0 0 1 1 1 2 0 0 0 0 1 3 1 1 2 1 0 0 1 0 1 3 3 2 0 0 1 1 0 2 2 3 2
1 0 0 0 1 1 1 1 1 0 0 0 0 1 1 1 1 2 0 0 1 1 0 1 1 3 2 0 0 0 0 1 1 1 1
2 0 0 1 0 0 1 1 1 1
;

```

To set up a multiple-group IRT model, you need to specify the grouping variable by using the **GROUP** statement. Very often you also want to specify cross-group equality constraints for different parameters. You can accomplish this by using the **EQUALITY** statement.

The model that is specified in the following statements is an extension of the model in [Example 53.1](#). The group variable, GroupVar, is specified in the **GROUP** statement. It has two values, 1 and 2, to indicate group membership. Equality constraints are specified in the **EQUALITY** statement.

```

proc irt data=IrtGroup;
  var item1-item8;
  group GroupVar;
  model item1-item4/resfunc=twop,
        item5-item8/resfunc=graded;
  equality item1-item4/parm=[intercept] between_gp=[1 2],
         _allgr_/parm=[slope] within_gp=[1];
run;

```

Two different sets of equality constraints have been specified. The first entry specifies equality constraints on the intercept parameters for item1 to item4 between group 1 and group 2:

```
item1-item4/parm=[intercept] between_gp=[1 2]
```

Notice that 1 and 2 are the actual values for the group variable GroupVar. The second entry specifies equality constraints on the slope parameters for all the graded response items within group 1:

```
_allgr_/parm=[slope] within_gp=[1]
```

[Output 53.3.1](#) shows the “Modeling Information” table and the “Group Information” table for this example. For multiple-group analysis, the “Modeling Information” table contains two extra pieces of information: the group variable and the number of groups. The “Group Information” table contains information about the data for each group. There are 272 observations that have been read and used for group 1; this number for group 2 is 328.

**Output 53.3.1** Modeling and Group Information

The IRT Procedure	
Modeling Information	
Data Set	WORK.IRTGROUP
Group Variable	GroupVar
Link Function	Logit
Number of Items	8
Number of Factors	1
Number of Groups	2
Number of Observations Read	600
Number of Observations Used	600
Estimation Method	Marginal Maximum Likelihood

**Output 53.3.1** *continued*

Group Information		
GroupVar	Nobs	
	Read	Used
1	272	272
2	328	328

Because there are two groups in this example, the IRT procedure produces two “Item Information” tables. For this example, these two tables contain the same information. That means that all the items have the same levels for the two groups. It is possible that the same item might have different numbers of levels, or maybe the same number of levels but different values. For example, an item has four levels, from 1 to 4, but one group might observe only levels 1 and 2, and the other group might observe only levels 3 and 4.

**Output 53.3.2** Item Information**The IRT Procedure**

Item Information GroupVar = 1			
Response Model	Item	Levels	Values
TwoP	item1	2	0 1
	item2	2	0 1
	item3	2	0 1
	item4	2	0 1
Graded	item5	2	0 1
	item6	3	1 2 3
	item7	3	1 2 3
	item8	3	1 2 3

Item Information GroupVar = 2			
Response Model	Item	Levels	Values
TwoP	item1	2	0 1
	item2	2	0 1
	item3	2	0 1
	item4	2	0 1
Graded	item5	2	0 1
	item6	3	1 2 3
	item7	3	1 2 3
	item8	3	1 2 3

Output 53.3.3 includes “Item Parameter Estimates” tables for both groups. You can see that the intercept parameters for item1 are the same for both groups. The same applies to item2 to item4. You can also see that the slope parameters have the same value for item5 to item8 in group 1. These results suggest that equality constraints that are specified in the **EQUALITY** statement have been fulfilled.

**Output 53.3.3** Parameter Estimates**The IRT Procedure**

Item Parameter Estimates GroupVar = 1					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
TwoP	item1	Difficulty	0.24545	0.07800	0.0008
		Slope	1.89309	0.37484	<.0001
	item2	Difficulty	0.49512	0.10112	<.0001
		Slope	2.00288	0.37954	<.0001
	item3	Difficulty	0.41490	0.10687	<.0001
		Slope	1.40433	0.26977	<.0001
	item4	Difficulty	0.57754	0.16188	0.0002
		Slope	0.90590	0.20213	<.0001
Graded	item5	Threshold	-0.71312	0.16145	<.0001
		Slope	0.96099	0.10872	<.0001
	item6	Threshold 1	-0.15789	0.14680	0.1411
		Threshold 2	1.42434	0.21483	<.0001
		Slope	0.96099	0.10872	<.0001
	item7	Threshold 1	-0.86494	0.16861	<.0001
		Threshold 2	0.81432	0.17160	<.0001
		Slope	0.96099	0.10872	<.0001
	item8	Threshold 1	-1.24131	0.19158	<.0001
		Threshold 2	0.37597	0.15203	0.0067
		Slope	0.96099	0.10872	<.0001

Item Parameter Estimates GroupVar = 2					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
TwoP	item1	Difficulty	0.31862	0.09331	0.0003
		Slope	1.45833	0.28819	<.0001
	item2	Difficulty	0.68599	0.12813	<.0001
		Slope	1.44561	0.28659	<.0001
	item3	Difficulty	0.51327	0.12276	<.0001
		Slope	1.13519	0.22854	<.0001
	item4	Difficulty	0.61629	0.16217	<.0001
		Slope	0.84894	0.18964	<.0001
Graded	item5	Threshold	-0.90646	0.27324	0.0005
		Slope	0.68749	0.17606	<.0001
	item6	Threshold 1	-0.77880	0.29844	0.0045
		Threshold 2	2.20837	0.57332	<.0001
		Slope	0.54126	0.14327	<.0001
	item7	Threshold 1	-0.84400	0.18217	<.0001
		Threshold 2	0.80547	0.16743	<.0001
		Slope	1.03946	0.18984	<.0001
	item8	Threshold 1	-1.18570	0.20677	<.0001
		Threshold 2	0.19876	0.12012	0.0490
		Slope	1.15609	0.21044	<.0001

## Example 53.4: Quality of Life Survey

The data set in this example comes from the 1978 Quality of American Life Survey. The survey was administered to a sample of US residents aged 18 years and older in 1978. Subjects were asked to rate their satisfaction with many different aspects of their lives. This example includes 14 items. Some of the items are as follows:

- satisfaction with community
- satisfaction with neighbors
- satisfaction with amount of education received
- satisfaction with health
- satisfaction with job
- satisfaction with income

Originally these items were designed with seven-point scales, where 1 indicates most unsatisfied and 7 indicates most satisfied. For illustration purposes, these items have been reorganized into a different number of categories, which ranges from 2 to 7. This example uses 1,000 random samples from the original data set. The following DATA step creates the data set `IrtQls`:

```
data IrtQls;
  input item1-item14 @@;
  datalines;
1  1  2  1  1  2  2  2  .  2  2  2  2  2
2  2  2  2  2  3  4  1  .  2  5  6  4  4

... more lines ...

1  1  1  1  2  2  2  2  .  1  1  1  1  3
;
```

By default, the IRT procedure uses the graded response model (GRM) and the logistic link for all the ordinal items and uses the two-parameter logistic model for all the binary items. In PROC IRT, you can specify different types of response models for different items by using the [MODEL](#) statement.

Because all the items in this example are designed to measure subjects' satisfaction with their lives, it is reasonable to start with a unidimensional IRT model. The following statements fit such a model by using the default model options:

```
ods graphics on;
proc irt data=IrtQls plots=(IIC TIC);
  var item1-item14;
run;
ods graphics off;
```

This example requests item information curves (IICs) and a test information curve (TIC) by using the `PLOTS=(IIC TIC)` option.

**Output 53.4.1** Eigenvalues of Polychoric Correlations**The IRT Procedure**

Eigenvalues of the Polychoric Correlation Matrix				
	Eigenvalue	Difference	Proportion	Cumulative
1	5.57173396	4.19614614	0.3980	0.3980
2	1.37558781	0.29273244	0.0983	0.4962
3	1.08285537	0.12600033	0.0773	0.5736
4	0.95685504	0.09108909	0.0683	0.6419
5	0.86576595	0.09758221	0.0618	0.7038
6	0.76818374	0.12571683	0.0549	0.7586
7	0.64246691	0.06108305	0.0459	0.8045
8	0.58138386	0.04214553	0.0415	0.8461
9	0.53923833	0.10092835	0.0385	0.8846
10	0.43830998	0.07346977	0.0313	0.9159
11	0.36484021	0.04667935	0.0261	0.9419
12	0.31816085	0.03905135	0.0227	0.9647
13	0.27910950	0.06360101	0.0199	0.9846
14	0.21550849		0.0154	1.0000

Output 53.4.1 shows the eigenvalue table for this example. You can see that the first eigenvalue is much greater than the others, suggesting that a unidimensional model is reasonable for the data.

In the context of item response theory, the amount of information that each item or the entire test provides might not be evenly distributed across the entire continuum of latent constructs. The value of the slope parameter indicates the amount of information that the item provides. For this example, parameter estimates and item information curves are shown in Output 53.4.2 and Output 53.4.3, respectively. By examining the parameter estimates and the item information curves, you can see that items that have high slope values have tall, narrow information curves. For example, because the slope value of item9 is much larger than the slope value of item1, the information curve is taller and narrower for item9 than it is for item1.

**Output 53.4.2** Parameter Estimates**The IRT Procedure**

Item Parameter Estimates					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
Graded	item1	Threshold 1	-2.09483	0.33839	<.0001
		Threshold 2	3.20986	0.50260	<.0001
		Slope	0.45899	0.07087	<.0001
	item2	Threshold 1	-0.54935	0.07526	<.0001
		Threshold 2	0.45190	0.07144	<.0001
		Slope	1.22585	0.09826	<.0001
	item5	Threshold 1	-0.71564	0.08937	<.0001
		Threshold 2	0.59637	0.08419	<.0001
		Slope	1.05329	0.08843	<.0001
	item6	Threshold 1	-0.60606	0.11597	<.0001
		Threshold 2	1.18041	0.15115	<.0001
		Slope	0.71308	0.07648	<.0001
	item7	Threshold 1	-0.78445	0.06313	<.0001
		Threshold 2	0.24069	0.05178	<.0001
		Threshold 3	0.87163	0.06538	<.0001
		Slope	1.90616	0.12225	<.0001
	item8	Threshold 1	-0.74471	0.07231	<.0001
		Threshold 2	0.60159	0.06732	<.0001


**The IRT Procedure**

Item Parameter Estimates					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
		Threshold 3	1.34996	0.09664	<.0001
		Slope	1.43137	0.09921	<.0001
	item10	Threshold 1	-0.33505	0.05731	<.0001
		Threshold 2	0.67167	0.06498	<.0001
		Slope	1.69974	0.12235	<.0001
	item11	Threshold 1	-1.02908	0.07590	<.0001
		Threshold 2	-0.03480	0.05564	0.2659
		Threshold 3	0.64199	0.06410	<.0001
		Threshold 4	1.34486	0.08998	<.0001
		Slope	1.67107	0.10914	<.0001
	item12	Threshold 1	-1.87573	0.13564	<.0001
		Threshold 2	-0.80823	0.08484	<.0001
		Threshold 3	-0.09862	0.06855	0.0751
		Threshold 4	0.59884	0.07678	<.0001
		Threshold 5	1.22786	0.10198	<.0001
		Threshold 6	1.83256	0.13657	<.0001
		Slope	1.19488	0.08609	<.0001
	item13	Threshold 1	-0.81329	0.05663	<.0001

**Output 53.4.2** *continued***The IRT Procedure**


Item Parameter Estimates					
Response Model	Item	Parameter	Estimate	Standard Error	Pr >  t
TwoP		Threshold 2	0.31547	0.04744	<.0001
		Threshold 3	1.07939	0.06425	<.0001
		Slope	2.49426	0.15672	<.0001
	item14	Threshold 1	-1.37128	0.09688	<.0001
		Threshold 2	0.35368	0.06259	<.0001
		Threshold 3	1.35631	0.09724	<.0001
		Slope	1.39958	0.09563	<.0001
	item3	Difficulty	-0.01166	0.09635	0.4519
		Slope	0.73660	0.08604	<.0001
	item4	Difficulty	-0.36507	0.06875	<.0001
		Slope	1.26289	0.11334	<.0001
	item9	Difficulty	0.16315	0.06378	0.0053
		Slope	1.89046	0.20555	<.0001


For individual items, most of the information concentrates around the area that is defined by the difficulty parameters. The binary response item provides most of the information around the difficulty parameter. For ordinal items, most of the information falls in the region between the lowest and the highest threshold parameters. By comparing the information curves for item7 and item9, you can also see that when response items have the same slope value, the ordinal item is more informative than the binary item.

**Output 53.4.3** Item Information Curves



Output 53.4.3 continued



**Output 53.4.4** Test Information Curves


When all items in a test are considered together, the information for measuring the latent trait is called the test information. Test information is computed as a summation of the information that is provided by all the items in the test. [Output 53.4.4](#) includes the test information curve for this example.

Item and test information are very useful for item selection. One important purpose of item selection is to maximize the test information across the continuum of latent construct of interest.

During the item selection process, ideally you want to select highly discriminating items whose threshold parameters cover the range of latent construct of interest. However, in practice you often encounter situations in which these highly discriminating items cannot provide enough information for a specific range of latent construct of interest, especially when these items are binary. In these situations, you might need to select some less discriminating items that can add information to the area that is not covered by these highly discriminating items.

For this example, the slope parameters range from 0.46 to 2.49, and the threshold parameters range from  $-2.1$  to  $3.2$ . Among these 14 items, three of them (item1, item3, and item6) have slope values less than 1. The slope value for item1 is less than 0.5, which is especially low. The item information curves suggest that these three items provide much less information than the other items. As a result, you might consider dropping these three items to economize future test administration. [Output 53.4.5](#) shows the test information curves for the original test, which has 14 items, and the shorter test, which excludes item1, item3, and item6. The two information curves are almost identical, suggesting that the shorter test provides almost the same amount of information as the longer test. Because the shorter test is more economical, it is preferred for future testing.

**Output 53.4.5** Test Information Curves



---

## References

- Bock, R. D. and Aitkin, M. (1981), "Marginal Maximum Likelihood Estimation of Item Parameters: Application of an EM Algorithm," *Psychometrika*, 46, 443–459.
- Bock, R. D. and Lieberman, M. (1970), "Fitting a Response Model for  $n$  Dichotomously Scored Items," *Psychometrika*, 35, 179–197.
- De Ayala, R. J. (2009), *The Theory and Practice of Item Response Theory*, New York: Guilford Press.
- Embretson, S. E. and Reise, S. P. (2000), *Item Response Theory for Psychologists*, Mahwah, NJ: Lawrence Erlbaum Associates.
- Lesaffre, E. and Spiessens, B. (2001), "On the Effect of the Number of Quadrature Points in a Logistic Random Effects Model: An Example," *Journal of the Royal Statistical Society, Series C*, 50, 325–335.
- McKinley, R. L. and Mills, C. N. (1985), "A Comparison of Several Goodness-of-Fit Statistics," *Applied Psychological Measurement*, 9, 49–57.
- Rabe-Hesketh, S., Skrondal, A., and Pickles, A. (2002), "Reliable Estimation of Generalized Linear Mixed Models Using Adaptive Quadrature," *Stata Journal*, 2, 1–21.
- Yen, W. M. (1981), "Using Simulation Results to Choose a Latent Trait Model," *Applied Psychological Measurement*, 5, 245–262.

# Subject Index

biquartimax method, [4020](#), [4021](#)

equamax method, [4020](#), [4021](#)

examples, IRT

    getting started, [4009](#)

IRT procedure, [4007](#)

    ODS table names, [4046](#)

    simplicity functions, [4020](#)

item response theory, [4008](#)

options summary

    PROC IRT statement, [4014](#)

parsimax method, [4020](#), [4021](#)

quartimax method, [4020](#), [4021](#)

quartimin method, [4021](#)

simplicity functions

    IRT procedure, [4020](#)

varimax method, [4020](#), [4021](#)



# Syntax Index

- ABSFCNV option
  - PROC IRT statement, [4015](#)
- ABSGCONV option
  - PROC IRT statement, [4015](#)
- ABSPCONV option
  - PROC IRT statement, [4015](#)
- BY statement
  - IRT procedure, [4022](#)
- COV statement, IRT procedure, [4022](#)
- DATA= option
  - PROC IRT statement, [4015](#)
- DESCENDING option
  - PROC IRT statement, [4015](#)
- EQUALITY statement
  - IRT procedure, [4026](#)
- FACTOR statement
  - IRT procedure, [4030](#)
- FCONV option
  - PROC IRT statement, [4016](#)
- FREQ statement
  - IRT procedure, [4034](#)
- GCONV= option
  - PROC IRT statement, [4016](#)
- GROUP statement
  - IRT procedure, [4034](#)
- IRT procedure, [4013](#)
  - syntax, [4013](#)
- IRT procedure, BY statement, [4022](#)
- IRT procedure, COV statement, [4022](#)
- IRT procedure, EQUALITY statement, [4026](#)
- IRT procedure, FACTOR statement, [4030](#)
- IRT procedure, FREQ statement, [4034](#)
- IRT procedure, GROUP statement, [4034](#)
- IRT procedure, MODEL statement, [4034](#)
- IRT procedure, PROC IRT statement, [4014](#)
  - ABSFCNV= option, [4015](#)
  - ABSGCONV= option, [4015](#)
  - ABSPCONV= option, [4015](#)
  - DATA= option, [4015](#)
  - DESCENDING option, [4015](#)
  - FCONV= option, [4016](#)
  - GCONV= option, [4016](#)
  - ITEMFIT option, [4016](#)
  - LINK= option, [4016](#)
  - MAXFUNC= option, [4016](#)
  - MAXITER= option, [4017](#)
  - MAXMITER= option, [4017](#)
  - NFACTOR= option, [4017](#)
  - NOAD option, [4017](#)
  - NOITPRINT option, [4017](#)
  - NOPRINT option, [4017](#)
  - OUT= option, [4018](#)
  - PINITIAL option, [4018](#)
  - PLOTS(UNPACK) option, [4019](#)
  - PLOTS= option, [4018](#)
  - POLYCHORIC option, [4018](#)
  - QPOINTS= option, [4019](#)
  - RCONVERGE= option, [4020](#)
  - RESFUNC= option, [4020](#)
  - RITER= option, [4020](#)
  - RORDER= option, [4021](#)
  - ROTATE= option, [4020](#)
  - SCOREMETHOD= option, [4021](#)
  - TECHNIQUE= option, [4017](#)
- IRT procedure, VAR statement, [4035](#)
- IRT procedure, VARIANCE statement, [4035](#)
- IRT procedure, WEIGHT statement, [4037](#)
- ITEMFIT option
  - PROC IRT statement, [4016](#)
- LINK= option
  - PROC IRT statement, [4016](#)
- MAXFUNC= option
  - PROC IRT statement, [4016](#)
- MAXITER= option
  - PROC IRT statement, [4017](#)
- MAXMITER= option
  - PROC IRT statement, [4017](#)
- MODEL statement
  - IRT procedure, [4034](#)
- NFACTOR= option
  - PROC IRT statement, [4017](#)
- NOAD option
  - PROC IRT statement, [4017](#)
- NOITPRINT option
  - PROC IRT statement, [4017](#)
- NOPRINT option
  - PROC IRT statement, [4017](#)

OUT= option  
PROC IRT statement, [4018](#)

PINITIAL option  
PROC IRT statement, [4018](#)

PLOTS= option  
PROC IRT statement, [4018](#)

POLYCHORIC option  
PROC IRT statement, [4018](#)

PROC IRT statement, *see* IRT procedure

QPOINTS= option  
PROC IRT statement, [4019](#)

RCONVERGE= option  
PROC IRT statement, [4020](#)

RESFUNC= option  
PROC IRT statement, [4020](#)

RITER= option  
PROC IRT statement, [4020](#)

RORDER= option  
PROC IRT statement, [4021](#)

ROTATE= option  
PROC IRT statement, [4020](#)

SCOREMETHOD= option  
PROC IRT statement, [4021](#)

TECHNIQUE= option  
PROC IRT statement, [4017](#)

UNPACK option  
PROC IRT statement, [4019](#)

VAR statement  
IRT procedure, [4035](#)

VARIANCE statement, IRT procedure, [4035](#)

WEIGHT statement  
IRT procedure, [4037](#)